Formal specification of the Cap9 kernel

Mikhail Mandrykin　　Ilya Shchepetkov

June 7, 2019

Contents

1 Introduction

This is an Isabelle/HOL theory that describes and proves the correctness of the Cap9 kernel specification.

2 Preliminaries

```
theory Cap9
imports
  "HOL−Word.Word"
  "HOL−Library.Adhoc_Overloading"
  "HOL−Library.DAList"
  "Word_Lib/Word_Lemmas"
begin
```

2.1 Type class instantiations

Instantiate *len* type class to extract lengths from word types avoiding repeated explicit numeric specification of the length e.g. *LENGTH(byte)* or *LENGTH('a :: len word)* instead of *8* or *LENGTH('a)*, where *'a* cannot be directly extracted from a type such as *'a word*.

```
instantiation word :: (len) len begin
definition len_word[simp]: "len_of (_ :: 'a::len word itself) = LENGTH('a)"
instance by (standard, simp)
end

lemma len_word': "LENGTH('a::len word) = LENGTH('a)" by (rule len_word)
```

Instantiate *size* type class for types of the form *'a itself*. This allows us to parametrize operations by word lengths using the dummy variables of type *'a word itself*. The operations cannot be directly parametrized by numbers as there is no lifting from term numbers to type numbers due to the lack of dependent types.

```
instantiation itself :: (len) size begin
definition size_itself where [simp, code]: "size (n::'a::len itself) = LENGTH('a)"
instance ..
end

declare unat_word_ariths[simp] word_size[simp] is_up_def[simp] wsst_TYs(1,2)[simp]
```

2.2 Word width

We introduce definition of the least numer of bits to hold the current value of a word. This is needed because in our specification we often word with *UCAST('a → 'b)*'ed values (right aligned subranges of bits), largely again due to the lack of dependent types (or true type-level functions), e.g. the it's hard to specify that the length of $a \bowtie b$ (where $\bowtie$ stands for concatenation) is the sum of the length of a and b, since length is a type parameter and there's no equivalent of sum on the type level. So we instead fix the length of $a \bowtie b$ to be the maximum possible one (say, 32 bytes) and then use conditions of the form *width* $a \leq s$ to specify that the actual "size" of a is s.

```
definition "width w ≡ LEAST n. unat w < 2 ^ n" for w :: "'a::len word"

lemma widthI[intro]: "⟦⋀ u. u < n ⟹ 2 ^ u ≤ unat w; unat w < 2 ^ n⟧ ⟹ width w = n"
  unfolding width_def Least_def
  using not_le
  apply (intro the_equality, blast)
  by (meson nat_less_le)

lemma width_wf[simp]: "∃! n. (∀ u < n. 2 ^ u ≤ unat w) ∧ unat w < 2 ^ n"
  (is "?Ex1 (unat w)")
proof (induction ("unat w"))
  case 0
  show "?Ex1 0" by (intro ex1I[of _ 0], auto)
next
  case (Suc x)
  then obtain n where x:"(∀u<n. 2 ^ u ≤ x) ∧ x < 2 ^ n " by auto
  show  "?Ex1 (Suc x)"
  proof (cases "Suc x < 2 ^ n")
    case True
    thus "?Ex1 (Suc x)"
      using x
      apply (intro ex1I[of _ "n"], auto)
      by (meson Suc_lessD leD linorder_neqE_nat)
  next
    case False
    thus "?Ex1 (Suc x)"
```

```
    using x
    apply (intro ex1I[of _ "Suc n"], auto simp add: less_Suc_eq)
    apply (intro antisym)
     apply (metis One_nat_def Suc_lessI Suc_n_not_le_n leI numeral_2_eq_2 power_increasing_iff)
    by (metis Suc_lessD le_antisym not_le not_less_eq_eq)
  qed
qed

lemma width_iff[iff]: "(width w = n) = ((∀ u < n. 2 ^ u ≤ unat w) ∧ unat w < 2 ^ n)"
  using width_wf widthI by metis

lemma width_le_size: "width x ≤ size x"
proof-
  {
    assume "size x < width x"
    hence "2 ^ size x ≤ unat x" using width_iff by metis
    hence "2 ^ size x ≤ uint x" unfolding unat_def by simp
  }
  thus ?thesis using uint_range_size[of x] by (force simp del:word_size)
qed

lemma width_le_size'[simp]: "size x ≤ n ⟹ width x ≤ n" by (insert width_le_size[of x], simp)

lemma nth_width_high[simp]: "width x ≤ i ⟹ ¬ x !! i"
proof (cases "i < size x")
  case False
  thus ?thesis by (simp add: test_bit_bin')
next
  case True
  hence "(x < 2 ^ i) = (unat x < 2 ^ i)"
    unfolding unat_def
    using word_2p_lem by fastforce
  moreover assume "width x ≤ i"
  then obtain n where "unat x < 2 ^ n" and "n ≤ i" using width_iff by metis
  hence "unat x < 2 ^ i"
    by (meson le_less_trans nat_power_less_imp_less not_less zero_less_numeral)
  ultimately show ?thesis using bang_is_le by force
qed

lemma width_zero[iff]: "(width x = 0) = (x = 0)"
proof
  show "width x = 0 ⟹ x = 0" using nth_width_high[of x] word_eq_iff[of x 0] nth_0 by (metis le0)
  show "x = 0 ⟹ width x = 0" by simp
qed

lemma width_zero'[simp]: "width 0 = 0" by simp

lemma width_one[simp]: "width 1 = 1" by simp

lemma high_zeros_less: "(∀ i ≥ u. ¬ x !! i) ⟹ unat x < 2 ^ u"
  (is "?high ⟹ _") for x :: "'a::len word"
proof-
  assume ?high
  have size:"size (mask u :: 'a word) = size x" by simp
  {
    fix i
    from ‹?high› have "(x AND mask u) !! i = x !! i"
      using nth_mask[of u i] size test_bit_size[of x i]
      by (subst word_ao_nth) (elim allE[of _ i], auto)
  }
```

```
  with ‹?high› have "x AND mask u = x" using word_eq_iff by blast
  thus ?thesis unfolding unat_def using mask_eq_iff by auto
qed

lemma nth_width_msb[simp]: "x ≠ 0 ⟹ x !! (width x - 1)"
proof (rule ccontr)
  fix x :: "'a word"
  assume "x ≠ 0"
  hence width:"width x > 0" using width_zero by fastforce
  assume "¬ x !! (width x - 1)"
  with width have "∀ i ≥ width x - 1. ¬ x !! i"
    using nth_width_high[of x] antisym_conv2 by fastforce
  hence "unat x < 2 ^ (width x - 1)" using high_zeros_less[of "width x - 1" x] by simp
  moreover from width have "unat x ≥ 2 ^ (width x - 1)" using width_iff[of x "width x"] by simp
  ultimately show False by simp
qed

lemma width_iff': "((∀ i > u. ¬ x !! i) ∧ x !! u) = (width x = Suc u)"
proof (rule; (elim conjE | intro conjI))
  assume "x !! u" and "∀ i > u. ¬ x !! i"
  show "width x = Suc u"
  proof (rule antisym)
    from ‹x !! u› show "width x ≥ Suc u" using not_less nth_width_high by force
    from ‹x !! u› have "x ≠ 0" by auto
    with ‹∀ i > u. ¬ x !! i› have "width x - 1 ≤ u" using not_less nth_width_msb by metis
    thus "width x ≤ Suc u" by simp
  qed
next
  assume "width x = Suc u"
  show "∀ i>u. ¬ x !! i" by (simp add:‹width x = Suc u›)
  from ‹width x = Suc u› show "x !! u" using nth_width_msb width_zero
    by (metis diff_Suc_1 old.nat.distinct(2))
qed

lemma width_word_log2: "x ≠ 0 ⟹ width x = Suc (word_log2 x)"
  using word_log2_nth_same word_log2_nth_not_set width_iff' test_bit_size
  by metis

lemma width_ucast[OF refl, simp]: "uc = ucast ⟹ is_up uc ⟹ width (uc x) = width x"
  by (metis uint_up_ucast unat_def width_def)

lemma width_ucast'[OF refl, simp]:
  "uc = ucast ⟹ width x ≤ size (uc x) ⟹ width (uc x) = width x"
proof-
  have "unat x < 2 ^ width x" unfolding width_def by (rule LeastI_ex, auto)
  moreover assume "width x ≤ size (uc x)"
  ultimately have "unat x < 2 ^ size (uc x)" by (simp add: less_le_trans)
  moreover assume "uc = ucast"
  ultimately have "unat x = unat (uc x)" by (metis unat_ucast mod_less word_size)
  thus ?thesis unfolding width_def by simp
qed

lemma width_lshift[simp]:
  "⟦x ≠ 0; n ≤ size x - width x⟧ ⟹ width (x << n) = width x + n"
  (is "⟦_; ?nbound⟧ ⟹ _")
proof-
  assume "x ≠ 0"
  hence 0:"width x = Suc (width x - 1)" using width_zero by (metis Suc_pred' neq0_conv)
  from ‹x ≠ 0› have 1:"width x > 0" by (auto intro:gr_zeroI)
  assume ?nbound
```

```
  {
    fix i
    from ‹?nbound› have "i ≥ size x ⟹ ¬ x !! (i − n)" by (auto simp add:le_diff_conv2)
    hence "(x << n) !! i = (n ≤ i ∧ x !! (i − n))" using nth_shiftl'[of x n i] by auto
  } note corr = this
  hence "∀ i > width x + n − 1. ¬ (x << n) !! i" by auto
  moreover from corr have "(x << n) !! (width x + n − 1)"
    using width_iff'[of "width x − 1" x] 1
    by auto
  ultimately have "width (x << n) = Suc (width x + n − 1)" using width_iff' by auto
  thus ?thesis using 0 by simp
qed

lemma width_lshift'[simp]: "n ≤ size x − width x ⟹ width (x << n) ≤ width x + n"
  using width_zero width_lshift shiftl_0 by (metis eq_iff le0)

lemma width_or[simp]: "width (x OR y) = max (width x) (width y)"
proof−
  {
    fix a b
    assume "width x = Suc a" and "width y = Suc b"
    hence "width (x OR y) = Suc (max a b)"
      using width_iff' word_ao_nth[of x y] max_less_iff_conj[of "a" "b"]
      by (metis (no_types) max_def)
  } note succs = this
  thus ?thesis
  proof (cases "width x = 0 ∨ width y = 0")
    case True
    thus ?thesis using width_zero word_log_esimps(3,9) by (metis max_0L max_0R)
  next
    case False
    with succs show ?thesis by (metis max_Suc_Suc not0_implies_Suc)
  qed
qed
```

2.3 Right zero-padding

Here's the first time we use *width*. If x is a value of size n right-aligned in a word of size $s = size\ x$ (note there's nowhere to keep the value n, since the size of x is some $s \geq n$, so we require it to be provided explicitly), then *rpad* n x will move the value x to the left. For the operation to be correct (no losing of significant higher bits) we need the precondition $width\ x \leq n$ in all the lemmas, hence the need for *width*.

```
definition rpad where "rpad n x ≡ x << size x − n"

lemma rpad_low[simp]: "⟦width x ≤ n; i < size x − n⟧ ⟹ ¬ (rpad n x) !! i"
  unfolding rpad_def by (simp add:nth_shiftl)

lemma rpad_high[simp]:
  "⟦width x ≤ n; n ≤ size x; size x − n ≤ i⟧ ⟹ (rpad n x) !! i = x !! (i + n − size x)"
  (is "⟦?xbound; ?nbound; i ≥ ?ibound⟧ ⟹ ?goal i")
proof−
  fix i
  assume ?xbound ?nbound and "i ≥ ?ibound"
  moreover from ‹?nbound› have "i + n − size x = i − ?ibound" by simp
  moreover from ‹?xbound› have "x !! (i + n − size x) ⟹ i < size x" by − (rule ccontr, simp)
  ultimately show "?goal i" unfolding rpad_def by (subst nth_shiftl', metis)
qed

lemma rpad_inj: "⟦width x ≤ n; width y ≤ n; n ≤ size x⟧ ⟹ rpad n x = rpad n y ⟹ x = y"
```

```
  (is "⟦?xbound; ?ybound; ?nbound; _⟧ ⟹ _")
  unfolding inj_def word_eq_iff
proof (intro allI impI)
  fix i
  let ?i' = "i + size x − n"
  assume ?xbound ?ybound ?nbound
  assume "∀j < LENGTH('a). rpad n x !! j = rpad n y !! j"
  hence "⋀ j. rpad n x !! j = rpad n y !! j" using test_bit_bin by blast
  from this[of ?i'] and ‹?xbound› ‹?ybound› ‹?nbound› show "x !! i = y !! i" by simp
qed
```

2.4 Spanning concatenation

```
abbreviation ucastl ("'(ucast')_ _" [1000, 100] 100) where
  "(ucast)_l a ≡ ucast a :: 'b word" for l :: "'b::len0 itself"

notation (input) ucastl ("'(ucast')_ _" [1000, 100] 100)

definition pad_join :: "'a::len word ⇒ nat ⇒ 'c::len itself ⇒ 'b::len word ⇒ 'c word"
  ("_ _◊_ _" [60, 1000, 1000, 61] 60) where
  "x _n◊_l y ≡ rpad n (ucast x) OR ucast y"

notation (input) pad_join ("_ _◊_ _" [60, 1000, 1000, 61] 60)

lemma pad_join_high:
  "⟦width a ≤ n; n ≤ size l; width b ≤ size l − n; size l − n ≤ i⟧
   ⟹ (a _n◊_l b) !! i = a !! (i + n − size l)"
  unfolding pad_join_def
  using nth_ucast nth_width_high by fastforce

lemma pad_join_high'[simp]:
  "⟦width a ≤ n; n ≤ size l; width b ≤ size l − n⟧ ⟹ a !! i = (a _n◊_l b) !! (i + size l − n)"
  using pad_join_high[of a n l b "i + size l − n"] by simp

lemma pad_join_mid[simp]:
  "⟦width a ≤ n; n ≤ size l; width b ≤ size l − n; width b ≤ i; i < size l − n⟧
   ⟹ ¬ (a _n◊_l b) !! i"
  unfolding pad_join_def by auto

lemma pad_join_low[simp]:
  "⟦width a ≤ n; n ≤ size l; width b ≤ size l − n; i < width b⟧ ⟹ (a _n◊_l b) !! i = b !! i"
  unfolding pad_join_def by (auto simp add: nth_ucast)

lemma pad_join_inj:
  assumes eq:"a _n◊_l b = c _n◊_l d"
  assumes a:"width a ≤ n" and c:"width c ≤ n"
  assumes n: "n ≤ size l"
  assumes b:"width b ≤ size l − n"
  assumes d:"width d ≤ size l − n"
  shows   "a = c" and "b = d"
proof−
  from eq have eq':"⋀j. (a _n◊_l b) !! j = (c _n◊_l d) !! j"
    using test_bit_bin unfolding word_eq_iff by auto
  moreover from a n b
  have "⋀ i. a !! i = (a _n◊_l b) !! (i + size l − n)" by simp
  moreover from c n d
  have "⋀ i. c !! i = (c _n◊_l d) !! (i + size l − n)" by simp
  ultimately show "a = c" unfolding word_eq_iff by auto

  {
    fix i
```

```
    from a n b have "i < width b ⟹ b !! i = (a n◇l b) !! i" by simp
    moreover from c n d have "i < width d ⟹ d !! i = (c n◇l d) !! i" by simp
    moreover have "i ≥ width b ⟹ ¬ b !! i" and "i ≥ width d ⟹ ¬ d !! i" by auto
    ultimately have "b !! i = d !! i"
      using eq'[of i] b d
        pad_join_mid[of a n l b i, OF a n b]
        pad_join_mid[of c n l d i, OF c n d]
      by (meson leI less_le_trans)
  }
  thus "b = d" unfolding word_eq_iff by simp
qed

lemma pad_join_inj'[dest!]:
 "⟦a n◇l b = c n◇l d;
   width a ≤ n; width c ≤ n; n ≤ size l;
   width b ≤ size l - n;
   width d ≤ size l - n⟧ ⟹ a = c ∧ b = d"
  apply (rule conjI)
  subgoal by (frule (4) pad_join_inj(1))
  by (frule (4) pad_join_inj(2))

lemma pad_join_and[simp]:
  assumes "width x ≤ n" "n ≤ m" "width a ≤ m" "m ≤ size l" "width b ≤ size l - m"
  shows   "(a m◇l b) AND rpad n x = rpad m a AND rpad n x"
  unfolding word_eq_iff
proof ((subst word_ao_nth)+, intro allI impI)
  from assms have 0:"n ≤ size x" by simp
  from assms have 1:"m ≤ size a" by simp
  fix i
  assume "i < LENGTH('a)"
  from assms show "((a m◇l b) !! i ∧ rpad n x !! i) = (rpad m a !! i ∧ rpad n x !! i)"
    using rpad_low[of x n i, OF assms(1)] rpad_high[of x n i, OF assms(1) 0]
          rpad_low[of a m i, OF assms(3)] rpad_high[of a m i, OF assms(3) 1]
          pad_join_high[of a m l b i, OF assms(3,4,5)]
          size_itself_def[of l] word_size[of x] word_size[of a]
    by (metis add.commute add_lessD1 le_Suc_ex le_diff_conv not_le)
qed
```

2.5 Deal with partially undefined results

```
definition restrict :: "'a::len word ⇒ nat set ⇒ 'a word" (infixl "↾" 60) where
 "restrict x s ≡ BITS i. i ∈ s ∧ x !! i"

lemma nth_restrict[iff]: "(x ↾ s) !! n = (n ∈ s ∧ x !! n)"
  unfolding restrict_def
  by (simp add: bang_conj_lt test_bit.eq_norm)

lemma restrict_inj2:
  assumes eq:"f x1 y1 OR v1 ↾ s = f x2 y2 OR v2 ↾ s"
  assumes fi:"⋀ x y i. i ∈ s ⟹ ¬ f x y !! i"
  assumes inj:"⋀ x1 y1 x2 y2. f x1 y1 = f x2 y2 ⟹ x1 = x2 ∧ y1 = y2"
  shows   "x1 = x2 ∧ y1 = y2"
proof-
  from eq and fi have "f x1 y1 = f x2 y2" unfolding word_eq_iff by auto
  with inj show ?thesis .
qed

lemmas restrict_inj_pad_join[dest] = restrict_inj2[of "λ x y. x _◇_ y"]
```

2.6 Plain concatenation

```
definition join :: "'a::len word ⇒ 'c::len itself ⇒ nat ⇒ 'b::len word ⇒ 'c word"
  ("_ _⋈_ _" [62,1000,1000,61] 61) where
  "(a l⋈n b) ≡ (ucast a << n) OR (ucast b)"

notation (input) join ("_ _⋈_ _" [62,1000,1000,61] 61)

lemma width_join:
  "⟦width a + n ≤ size l; width b ≤ n⟧ ⟹ width (a l⋈n b) ≤ width a + n"
  (is "⟦?abound; ?bbound⟧ ⟹ _")
proof–
  assume ?abound and ?bbound
  moreover hence "width b ≤ size l" by simp
  ultimately show ?thesis
    using width_lshift'[of n "(ucast)l a"]
    unfolding join_def
    by simp
qed

lemma width_join'[simp]:
  "⟦width a + n ≤ size l; width b ≤ n; width a + n ≤ q⟧ ⟹ width (a l⋈n b) ≤ q"
  by (drule (1) width_join, simp)

lemma join_high[simp]:
  "⟦width a + n ≤ size l; width b ≤ n; width a + n ≤ i⟧ ⟹ ¬ (a l⋈n b) !! i"
  by (drule (1) width_join, simp)

lemma join_mid:
  "⟦width a + n ≤ size l; width b ≤ n; n ≤ i; i < width a + n⟧ ⟹ (a l⋈n b) !! i = a !! (i − n)"
  apply (subgoal_tac "i < size ((ucast)l a) ∧ size ((ucast)l a) = size l")
  unfolding join_def
  using word_ao_nth nth_ucast nth_width_high nth_shiftl'
   apply (metis less_imp_diff_less order_trans word_size)
  by simp

lemma join_mid'[simp]:
  "⟦width a + n ≤ size l; width b ≤ n⟧ ⟹ a !! i = (a l⋈n b) !! (i + n)"
  using join_mid[of a n l b "i + n"] nth_width_high[of a i] join_high[of a n l b "i + n"]
  by force

lemma join_low[simp]:
  "⟦width a + n ≤ size l; width b ≤ n; i < n⟧ ⟹ (a l⋈n b) !! i = b !! i"
  unfolding join_def
  by (simp add: nth_shiftl nth_ucast)

lemma join_inj:
  assumes eq:"a l⋈n b = c l⋈n d"
  assumes "width a + n ≤ size l" and "width b ≤ n"
  assumes "width c + n ≤ size l" and "width d ≤ n"
  shows   "a = c" and "b = d"
proof–
  from assms show "a = c" unfolding word_eq_iff using join_mid' eq by metis
  from assms show "b = d" unfolding word_eq_iff using join_low nth_width_high
    by (metis eq less_le_trans not_le)
qed

lemma join_inj'[dest!]:
  "⟦a l⋈n b = c l⋈n d;
    width a + n ≤ size l; width b ≤ n;
    width c + n ≤ size l; width d ≤ n⟧ ⟹ a = c ∧ b = d"
  apply (rule conjI)
```

```
  subgoal by (frule (4) join_inj(1))
  by (frule (4) join_inj(2))

lemma join_and:
  assumes "width x ≤ n" "n ≤ size l" "k ≤ size l" "m ≤ k"
          "n ≤ k - m" "width a ≤ k - m" "width a + m ≤ k" "width b ≤ m"
  shows   "rpad k (a ⌊⋈m b) AND rpad n x = rpad (k - m) a AND rpad n x"
  unfolding word_eq_iff
proof ((subst word_ao_nth)+, intro allI impI)
  from assms have 0:"n ≤ size x" by simp
  from assms have 1:"k - m ≤ size a" by simp
  from assms have 2:"width (a ⌊⋈m b) ≤ k" by simp
  from assms have 3:"k ≤ size (a ⌊⋈m b)" by simp
  from assms have 4:"width a + m ≤ size l" by simp
  fix i
  assume "i < LENGTH('a)"
  moreover with assms have "i + k - size (a ⌊⋈m b) - m = i + (k - m) - size a" by simp
  moreover from assms have "i + k - size (a ⌊⋈m b) < m ⟹ i < size x - n" by simp
  moreover from assms have
    "⟦i ≥ size l - k; m ≤ i + k - size (a ⌊⋈m b)⟧ ⟹ size a - (k - m) ≤ i" by simp
  moreover from assms have "width a + m ≤ i + k - size (a ⌊⋈m b) ⟹ ¬ rpad (k - m) a !! i"
    by (simp add: nth_shiftl' rpad_def)
  moreover from assms have "¬ i ≥ size l - k ⟹ i < size x - n" by simp
  ultimately show "(rpad k (a ⌊⋈m b) !! i ∧ rpad n x !! i) =
                  (rpad (k - m) a !! i ∧ rpad n x !! i)"
    using assms
          rpad_high[of x n i, OF assms(1) 0] rpad_low[of x n i, OF assms(1)]
          rpad_high[of a "k - m" i, OF assms(6) 1] rpad_low[of a "k - m" i, OF assms(6)]
          rpad_high[of "a ⌊⋈m b" k i, OF 2 3] rpad_low[of "a ⌊⋈m b" k i, OF 2]
          join_high[of a m l b "i + k - size (a ⌊⋈m b)", OF 4 assms(8)]
          join_mid[of a m l b "i + k - size (a ⌊⋈m b)", OF 4 assms(8)]
          join_low[of a m l b "i + k - size (a ⌊⋈m b)", OF 4 assms(8)]
          size_itself_def[of l] word_size[of x] word_size[of a] word_size[of "a ⌊⋈m b"]
    by (metis not_le)
qed

lemma join_and'[simp]:
   "⟦width x ≤ n; n ≤ size l; k ≤ size l; m ≤ k;
     n ≤ k - m; width a ≤ k - m; width a + m ≤ k; width b ≤ m⟧ ⟹
    rpad k (a ⌊⋈m b) AND rpad n x = rpad (k - m) (ucast a) AND rpad n x"
  using join_and[of x n l k m "ucast a" b] unfolding join_def
  by (simp add: ucast_id)
```

3 Data formats

3.1 Procedure keys

Procedure keys are represented as 24-byte (192 bits) machine words.

```
type_synonym word24 = "192 word" — 24 bytes
type_synonym key = word24
```

3.2 Storage state

Byte is 8-bit machine word:

```
type_synonym byte = "8 word"
```

32-byte machine words that are used to model keys and values of the storage.

```
type_synonym word32 = "256 word" — 32 bytes
```

Storage is a function that takes a 32-byte word (key) and returns another 32-byte word (value).

type_synonym *storage* = "*word32* ⇒ *word32*"

3.3 Common notation

Specialize previously defined general concatenation operations for the fixed result size of 32 bytes. Thus we avoid lots of redundant type annotations for every intermediate result (note that these intermediate types cannot be inferred automatically (in a purely Hindley-Milner setting as in Isabelle), because this would require type-level functions/dependent types).

```
abbreviation "len (_ :: 'a::len word itself) ≡ TYPE('a)"

no_notation join  ("_ _⋈_ _" [62,1000,1000,61] 61)
no_notation (input) join ("_ _⋈_ _" [62,1000,1000,61] 61)

abbreviation join32 ("_ ⋈_ _" [62,1000,61] 61) where
  "a ⋈n b ≡ join a (len TYPE(word32)) (n ∗ 8) b"
abbreviation (output) join32_out ("_ ⋈_ _" [62,1000,61] 61) where
  "join32_out a n b ≡ join a (TYPE(256)) n b"
notation (input) join32 ("_ ⋈_ _" [62,1000,61] 61)

no_notation pad_join  ("_ _◊_ _" [60,1000,1000,61] 60)
no_notation (input) pad_join ("_ _◊_ _" [60,1000,1000,61] 60)

abbreviation pad_join32 ("_ _◊ _" [60,1000,61] 60) where
  "a n◊ b ≡ pad_join a (n ∗ 8) (len TYPE(word32)) b"
abbreviation (output) pad_join32_out ("_ _◊ _" [60,1000,61] 60) where
  "pad_join32_out a n b ≡ pad_join a n (TYPE(256)) b"
notation (input) pad_join32 ("_ _◊ _" [60,1000,61] 60)
```

Override treatment of hexidecimal numeric constants to make them monomorphic words of fixed length, mimicking the notation used in the informal specification (e.g. *1::'a*) is always a word 1 byte long and is not, say, the natural number one). Otherwise, again, lots of redundant type annotations would arise.

```
parse_ast_translation ‹
  let
    open Ast
    fun mk_numeral t = mk_appl (Constant @{syntax_const _Numeral}) t
    fun mk_word_numeral num t =
      if String.isPrefix 0x num then
        mk_appl (Constant @{syntax_const _constrain})
          [mk_numeral t,
           mk_appl (Constant @{type_syntax word})
             [mk_appl (Constant @{syntax_const _NumeralType})
             [Variable (4 ∗ (size num − 2) |> string_of_int)]]]
      else
        mk_numeral t
    fun numeral_ast_tr ctxt (t as [Appl [Constant @{syntax_const _constrain},
                                         Constant num,
                                         _]])
                                                = mk_word_numeral num t
      | numeral_ast_tr ctxt (t as [Constant num]) = mk_word_numeral num t
      | numeral_ast_tr _ t                         = mk_numeral t
      | numeral_ast_tr _ t                         = raise AST (@{syntax_const _Numeral}, t)
  in
    [(@{syntax_const _Numeral}, numeral_ast_tr)]
  end
›
```

Introduce generic notation for representation/encoding of various "logical"/abstract entities into machine words. We use adhoc overloading to use the same notation for various types of entities (indices, offsets, addresses, capabilities etc.).

3.4 Addresses

```
no_notation floor ("⌊_⌋")

consts rep :: "'a ⇒ 'b" ("⌊_⌋")

no_notation ceiling ("⌈_⌉")

consts abs :: "'a ⇒ 'b" ("⌈_⌉")

definition "maybe_inv f y ≡ if y ∈ range f then Some (the_inv f y) else None"

lemma maybe_inv_inj[intro]: "inj f ⟹ maybe_inv f (f x) = Some x"
  unfolding maybe_inv_def
  by (auto simp add:inj_def the_inv_f_f)

lemma maybe_inv_inj'[dest]: "⟦inj f; maybe_inv f y = Some x⟧ ⟹ f x = y"
  unfolding maybe_inv_def
  by (auto intro:f_the_inv_into_f simp add:inj_def split:if_splits)

locale invertible =
  fixes rep :: "'a ⇒ 'b" ("⌊_⌋")
  assumes inj:"inj rep"
begin
definition inv :: "'b ⇒ 'a option" where "inv ≡ maybe_inv rep"

lemmas inv_inj[folded inv_def, simp] = maybe_inv_inj[OF inj]

lemmas inv_inj'[folded inv_def, simp] = maybe_inv_inj'[OF inj]
end

definition "range2 f ≡ {y. ∃x₁ ∈ UNIV. ∃ x₂ ∈ UNIV. y = f x₁ x₂}"

definition "the_inv2 f ≡ λ x. THE y. ∃ y'. f y y' = x"

definition "maybe_inv2 f y ≡ if y ∈ range2 f then Some (the_inv2 f y) else None"

definition "inj2 f ≡ ∀ x₁ x₂ y₁ y₂. f x₁ y₁ = f x₂ y₂ ⟶ x₁ = x₂"

lemma inj2I: "(⋀ x₁ x₂ y₁ y₂. f x₁ y₁ = f x₂ y₂ ⟹ x₁ = x₂) ⟹ inj2 f" unfolding inj2_def
  by blast

lemma maybe_inv2_inj[intro]: "inj2 f ⟹ maybe_inv2 f (f x y) = Some x"
  unfolding maybe_inv2_def the_inv2_def inj2_def range2_def
  by (simp split:if_splits, blast)

lemma maybe_inv2_inj'[dest]:
  "⟦inj2 f; maybe_inv2 f y = Some x⟧ ⟹ ∃ y'. f x y' = y"
  unfolding maybe_inv2_def the_inv2_def range2_def inj2_def
  by (force split:if_splits intro:theI)

locale invertible2 =
  fixes rep :: "'a ⇒ 'b ⇒ 'c" ("⌊_⌋")
  assumes inj:"inj2 rep"
begin
definition inv2 :: "'c ⇒ 'a option" where "inv2 ≡ maybe_inv2 rep"
```

```
lemmas inv2_inj[folded inv2_def, simp] = maybe_inv2_inj[OF inj]

lemmas inv2_inj'[folded inv_def, simp] = maybe_inv2_inj'[OF inj]
end
```

We don't include *Null* capability into the type. It is only handled specially inside the call delegation, otherwise it only complicates the proofs with side conditions $\neq$ *Null*. So there will be separate type *call* defined as *capability option* to respect the fact that it can be *Null*.

In general, in the following we strive to make all encoding functions injective without any preconditions. All the necessary invariants are built into the type definitions.

```
datatype capability =
    Call
  | Reg
  | Del
  | Entry
  | Write
  | Log
  | Gas

definition cap_type_rep :: "capability ⇒ byte" where
  "cap_type_rep c ≡ case c of
      Call  ⇒ 0x03
    | Reg   ⇒ 0x04
    | Del   ⇒ 0x05
    | Entry ⇒ 0x06
    | Write ⇒ 0x07
    | Log   ⇒ 0x08
    | Gas   ⇒ 0x09"

adhoc_overloading rep cap_type_rep

lemma cap_type_rep_rng[simp]: "⌊c⌋ ∈ {0x03..0x09}" for c :: capability
  unfolding cap_type_rep_def by (simp split:capability.split)

lemma cap_type_rep_inj[simp]: "⌊c₁⌋ = ⌊c₂⌋ ⟹ c₁ = c₂" for c₁ c₂ :: capability
  unfolding cap_type_rep_def
  by (simp split:capability.splits)

lemma width_cap_type: "width (⌊c⌋+ 1) ≤ 4" for c :: capability
proof (rule ccontr, drule not_le_imp_less)
  assume "4 < width (⌊c⌋ + 1)"
  moreover hence "(⌊c⌋ + 1) !! (width (⌊c⌋ + 1) − 1)" using nth_width_msb by force
  ultimately obtain n where "(⌊c⌋ + 1) !! n" and "n ≥ 4" by (metis le_step_down_nat nat_less_le)
  thus False unfolding cap_type_rep_def by (simp split:capability.splits)
qed

lemma width_cap_type'[simp]: "4 ≤ n ⟹ width (⌊c⌋ + 1) ≤ n" for c :: capability
  using width_cap_type[of c] by simp

lemma cap_type_nonzero[simp]: "⌊c⌋ ≠ 0" for c :: capability
  unfolding cap_type_rep_def by (simp split:capability.splits)

typedef capability_index = "{i :: nat. i < 2 ^ LENGTH(byte) − 1}"
  morphisms cap_index_rep' cap_index'
  by (intro exI[of _ "0"], simp)

adhoc_overloading rep cap_index_rep'

definition "cap_index_rep i ≡ of_nat (⌊i⌋ + 1) :: byte" for i :: capability_index
```

```
adhoc_overloading rep cap_index_rep

lemma width_cap_index: "width ⌊i⌋ ≤ 8" for i :: capability_index by simp

lemma width_cap_index'[simp]: "8 ≤ n ⟹ width (⌊i⌋) ≤ n" for i :: capability_index by simp

lemma cap_index_nonzero[simp]: "⌊i⌋ ≠ 0x00" for i :: capability_index
  unfolding cap_index_rep_def using cap_index_rep'[of i] of_nat_neq_0[of "Suc ⌊i⌋"]
  by force

lemma cap_index_inj[simp]: "(⌊i₁⌋ :: byte) = ⌊i₂⌋ ⟹ i₁ = i₂" for i₁ i₂ :: capability_index
  unfolding cap_index_rep_def
  using cap_index_rep'[of i₁] cap_index_rep'[of i₂] word_of_nat_inj[of "⌊i₁⌋" "⌊i₂⌋"]
        cap_index_rep'_inject
  by force

lemmas cap_index_invertible[intro] = invertible.intro[OF injI, OF cap_index_inj]

interpretation cap_index_inv: invertible cap_index_rep ..

adhoc_overloading abs cap_index_inv.inv

type_synonym capability_offset = byte

datatype data_offset =
  Addr
  | Index
  | Ncaps capability
  | Cap capability capability_index capability_offset

definition data_offset_rep :: "data_offset ⇒ word32" where
 "data_offset_rep off ≡ case off of
    Addr          ⇒ 0x00  ⋈₂ 0x00  ⋈₁  0x00
  | Index         ⇒ 0x00  ⋈₂ 0x00  ⋈₁  0x01
  | Ncaps ty      ⇒ ⌊ty⌋  ⋈₂ 0x00  ⋈₁  0x00
  | Cap ty i off  ⇒ ⌊ty⌋  ⋈₂ ⌊i⌋    ⋈₁  off"

adhoc_overloading rep data_offset_rep

lemma data_offset_inj[simp]:
  "⌊d₁⌋ = ⌊d₂⌋ ⟹ d₁ = d₂" for d₁ d₂ :: data_offset
  unfolding data_offset_rep_def
  by (auto split:data_offset.splits)

lemma width_data_offset: "width ⌊d⌋ ≤ 3 ∗ 8" for d :: data_offset
  unfolding data_offset_rep_def
  by (simp split:data_offset.splits)

lemma width_data_offset'[simp]: "3 ∗ 8 ≤ n ⟹ width ⌊d⌋ ≤ n" for d :: data_offset
  using width_data_offset[of d] by simp

typedef key_index = "{i :: nat. i < 2 ˆ LENGTH(key) − 1}" morphisms key_index_rep' key_index
  by (rule exI[of _ "0"], simp)

adhoc_overloading rep key_index_rep'

datatype address =
   Heap_proc key data_offset
  | Nprocs
  | Proc_key key_index
```

```
  | Kernel
  | Curr_proc
  | Entry_proc

definition "key_index_rep i ≡ of_nat (⌊i⌋ + 1) :: key" for i :: key_index

adhoc_overloading rep key_index_rep

lemma key_index_nonzero[simp]: "⌊i⌋ ≠ (0 :: key)" for i :: key_index
  unfolding key_index_rep_def using key_index_rep'[of i]
  by (intro of_nat_neq_0, simp_all)

lemma key_index_inj[simp]: "(⌊i1⌋ :: key) = ⌊i2⌋ ⟹ i1 = i2" for i :: key_index
  unfolding key_index_rep_def using key_index_rep'[of i1] key_index_rep'[of i2]
  by (simp add:key_index_rep'_inject of_nat_inj)

abbreviation "kern_prefix ≡ 0xffffffff"

definition addr_rep :: "address ⇒ word32" where
  "addr_rep a ≡ case a of
    Heap_proc k offs ⇒ kern_prefix ⋈1 0x00 5◊ k          ⋈3 ⌊offs⌋
  | Nprocs        ⇒ kern_prefix ⋈1 0x01 5◊ (0 :: key) ⋈3 0x000000
  | Proc_key i    ⇒ kern_prefix ⋈1 0x01 5◊ ⌊i⌋         ⋈3 0x000000
  | Kernel        ⇒ kern_prefix ⋈1 0x02 5◊ (0 :: key) ⋈3 0x000000
  | Curr_proc     ⇒ kern_prefix ⋈1 0x03 5◊ (0 :: key) ⋈3 0x000000
  | Entry_proc    ⇒ kern_prefix ⋈1 0x04 5◊ (0 :: key) ⋈3 0x000000"

adhoc_overloading rep addr_rep

lemma addr_inj[simp]: "⌊a1⌋ = ⌊a2⌋ ⟹ a1 = a2" for a1 a2 :: address
  unfolding addr_rep_def
  by (split address.splits) (force split:address.splits)+

lemmas addr_invertible[intro] = invertible.intro[OF injI, OF addr_inj]

interpretation addr_inv: invertible addr_rep ..

adhoc_overloading abs addr_inv.inv

abbreviation "prefix_bound ≡ rpad (size kern_prefix) (ucast kern_prefix :: word32)"

lemma prefix_bound: "unat prefix_bound < 2 ^ LENGTH(word32)" unfolding rpad_def by simp

lemma prefix_bound'[simplified, simp]: "x ≤ unat prefix_bound ⟹ x < 2 ^ LENGTH(word32)"
  using prefix_bound by simp

lemma addr_prefix[intro]: "limited_and prefix_bound ⌊a⌋" for a :: address
  unfolding limited_and_def addr_rep_def
  by (subst word_bw_comms) (auto split:address.split simp del:ucast_bintr)
```

3.5 Capability formats

```
no_notation abs ("⌈_⌉")

locale cap_sub =
  fixes set_of :: "'a ⇒ 'b set" ("⌈_⌉")
  fixes sub :: "'a ⇒ 'a ⇒ bool" ("(_/ ⊆c _)" [51, 51] 50)
  assumes wd:"a ⊆c b = (⌈a⌉ ⊆ ⌈b⌉)" begin

lemma sub_refl: "a ⊆c a" using wd by auto
```

```
lemma sub_trans: "⟦a ⊆c b; b ⊆c c⟧ ⟹ a ⊆c c" using wd by blast
end

notation abs ("⌈_⌉")

consts sub :: "'a ⇒ 'a ⇒ bool" ("(_/ ⊆c _)" [51, 51] 50)
```

3.5.1 Call, Register and Delete capabilities

```
typedef prefix_size = "{n :: nat. n ≤ LENGTH(key)}"
  morphisms prefix_size_rep' prefix_size
  by auto

adhoc_overloading rep prefix_size_rep'

definition "prefix_size_rep s ≡ of_nat ⌊s⌋ :: byte" for s :: prefix_size

adhoc_overloading rep prefix_size_rep

lemma prefix_size_inj[simp]: "(⌊s1⌋ :: byte) = ⌊s2⌋ ⟹ s1 = s2" for s1 s2 :: prefix_size
  unfolding prefix_size_rep_def using prefix_size_rep'[of s1] prefix_size_rep'[of s2]
  by (simp add:prefix_size_rep'_inject of_nat_inj)

lemma prefix_size_rep_less[simp]: "LENGTH(key) ≤ n ⟹ ⌊s⌋ ≤ (n :: nat)" for s :: prefix_size
  using prefix_size_rep'[of s] by simp

type_synonym prefixed_capability = "prefix_size × key"

definition
  "set_of_pref_cap sk ≡ let (s, k) = sk in {k' :: key. take ⌊s⌋ (to_bl k') = take ⌊s⌋ (to_bl k)}"
  for sk :: prefixed_capability

adhoc_overloading abs set_of_pref_cap

definition "pref_cap_sub A B ≡
  let (sA, kA) = A in let (sB, kB) = B in
  (⌊sA⌋ :: nat) ≥ ⌊sB⌋ ∧ take ⌊sB⌋ (to_bl kA) = take ⌊sB⌋ (to_bl kB)"
  for A B :: prefixed_capability

adhoc_overloading sub pref_cap_sub

lemma nth_take_i[dest]: "⟦take n a = take n b; i < n⟧ ⟹ a ! i = b ! i"
  by (metis nth_take)

lemma take_less_diff:
  fixes l' l'' :: "'a list"
  assumes ex:"⋀ u :: 'a. ∃ u'. u' ≠ u"
  assumes "n < m"
  assumes "length l' = length l''"
  assumes "n ≤ length l'"
  assumes "m ≤ length l'"
  obtains l where
      "length l = length l'"
  and "take n l = take n l'"
  and "take m l ≠ take m l''"
proof-
  let ?x = "l'' ! n"
  from ex obtain y where neq:"y ≠ ?x" by auto
  let ?l = "take n l' @ y # drop (n + 1) l'"
  from assms have 0:"n = length (take n l') + 0" by simp
  from assms have "take n ?l = take n l'" by simp
```

```
  moreover from assms and neq have "take m ?l ≠ take m l''"
    using 0 nth_take_i nth_append_length
    by (metis add.right_neutral)
  moreover have "length ?l = length l'" using assms by auto
  ultimately show ?thesis using that by blast
qed

lemma pref_cap_sub_iff[iff]: "a ⊆c b = (⌈a⌉ ⊆ ⌈b⌉)" for a b :: prefixed_capability
proof
  show "a ⊆c b ⟹ ⌈a⌉ ⊆ ⌈b⌉"
    unfolding pref_cap_sub_def set_of_pref_cap_def
    by (force intro:nth_take_lemma)
  {
    fix n m :: prefix_size
    fix x y :: key
    assume "⌊n⌋ < (⌊m⌋ :: nat)"
    then obtain z where
      "length z = size x"
      "take ⌊n⌋ z = take ⌊n⌋ (to_bl x)" and "take ⌊m⌋ z ≠ take ⌊m⌋ (to_bl y)"
      using take_less_diff[of "⌊n⌋" "⌊m⌋" "to_bl x" "to_bl y"]
      by auto
    moreover hence "to_bl (of_bl z :: key) = z" by (intro word_bl.Abs_inverse[of z], simp)
    ultimately
    have "∃ u :: key.
           take ⌊n⌋ (to_bl u) = take ⌊n⌋ (to_bl x) ∧ take ⌊m⌋ (to_bl u) ≠ take ⌊m⌋ (to_bl y)"
      by metis
  }
  thus "⌈a⌉ ⊆ ⌈b⌉ ⟹ a ⊆c b"
    unfolding pref_cap_sub_def set_of_pref_cap_def subset_eq
    apply (auto split:prod.split)
    by (erule contrapos_pp[of "∀ x. _ x"], simp)
qed

lemmas pref_cap_subsets[intro] = cap_sub.intro[OF pref_cap_sub_iff]

interpretation pref_cap_sub: cap_sub set_of_pref_cap pref_cap_sub ..

definition "pref_cap_rep sk r ≡
  let (s, k) = sk in ⌊s⌋ 1◊ k OR r ↾ {LENGTH(key) + 1 ..<LENGTH(word32) − LENGTH(byte)}"
  for sk :: prefixed_capability

adhoc_overloading rep pref_cap_rep

lemma pref_cap_rep_inj_helper_inj[simp]: "⌊s1⌋ 1◊ k1 = ⌊s2⌋ 1◊ k2 ⟹ s1 = s2 ∧ k1 = k2"
  for s1 s2 :: prefix_size and k1 k2 :: key
  by auto

lemma pref_cap_rep_inj_helper_zero[simplified, simp]:
  "n ∈ {LENGTH(key) + 1 ..<LENGTH(word32) − LENGTH(byte)} ⟹ ¬ (⌊s⌋ 1◊ k) !! n"
  for s :: prefix_size and k :: key
  by simp

lemma pref_cap_rep_inj[simp]: "⌊c1⌋ r1 = ⌊c2⌋ r2 ⟹ c1 = c2" for c1 c2 :: prefixed_capability
  unfolding pref_cap_rep_def
  by (auto split:prod.splits)

lemmas pref_cap_invertible[intro] = invertible2.intro[OF inj2I, OF pref_cap_rep_inj]

interpretation pref_cap_inv: invertible2 pref_cap_rep ..
```

```
adhoc_overloading abs pref_cap_inv.inv2
```

3.5.2 Write capability

```
typedef write_capability = "{(a :: word32, n). n < unat prefix_bound − unat a}"
  morphisms write_cap_rep' write_cap
  unfolding rpad_def
  by (intro exI[of _ "(0, 0)"], simp)

adhoc_overloading rep write_cap_rep'

lemma write_cap_additional_bound[simplified, simp]:
  "snd ⌊w⌋ < unat prefix_bound" for w :: write_capability
  using write_cap_rep'[of w]
  by (auto split:prod.split)

lemma write_cap_additional_bound'[simplified, simp]:
  "unat prefix_bound ≤ n ⟹ ⌊w⌋ = (a, b) ⟹ b < n"
  using write_cap_additional_bound[of w] by simp

lemma write_cap_bound: "unat (fst ⌊w⌋) + snd ⌊w⌋ < unat prefix_bound"
  using write_cap_rep'[of w]
  by (simp split:prod.splits)

lemma write_cap_bound'[simplified, simp]: "⌊w⌋ = (a, b) ⟹ unat a + b < unat prefix_bound"
  using write_cap_bound[of w] by simp

lemma write_cap_no_overflow: "fst ⌊w⌋ ≤ fst ⌊w⌋ + of_nat (snd ⌊w⌋)" for w :: write_capability
  by (simp add:word_le_nat_alt unat_of_nat_eq less_imp_le)

lemma write_cap_no_overflow'[simp]: "⌊w⌋ = (a, b) ⟹ a ≤ a + of_nat b"
  for w :: write_capability
  using write_cap_no_overflow[of w] by simp

lemma nth_kern_prefix: "kern_prefix !! i = (i < size kern_prefix)"
proof−
  fix i
  {
    fix c :: nat
    assume "i < c"
    then consider "i = c − 1" | "i < c − 1 ∧ c ≥ 1"
      by fastforce
  } note elim = this
  have "i < size kern_prefix ⟹ kern_prefix !! i"
    by (subst test_bit_bl, (erule elim, simp_all)+)
  moreover have "i ≥ size kern_prefix ⟹ ¬ kern_prefix !! i" by simp
  ultimately show "kern_prefix !! i = (i < size kern_prefix)" by auto
qed

lemma nth_prefix_bound[iff]:
  "prefix_bound !! i = (i ∈ {LENGTH(word32) − size (kern_prefix)..<LENGTH(word32)})"
  (is "_ = (i ∈ {?l..<?r})")
proof−
  have 0:"is_up (ucast :: 32 word ⇒ word32)" by simp
  have 1:"width (ucast kern_prefix :: word32) ≤ size kern_prefix"
    using width_ucast[of kern_prefix, OF 0] by (simp del:width_iff)
  fix i
  show "prefix_bound !! i = (i ∈ {?l..<?r})"
    using rpad_high
      [of "(ucast)(len TYPE(word32)) kern_prefix" "size (kern_prefix)" i, OF 1, simplified]
```

```
      rpad_low
      [of "(ucast)(len TYPE(word32)) kern_prefix" "size (kern_prefix)" i, OF 1, simplified]
      nth_kern_prefix[of "i − ?l", simplified] nth_ucast[of kern_prefix i, simplified]
      test_bit_size[of prefix_bound i, simplified]
  by (simp (no_asm_simp)) linarith
qed

lemma write_cap_high[dest]:
  "unat a < unat prefix_bound ⟹
   ∃ i ∈ {LENGTH(word32) − size (kern_prefix)..<LENGTH(word32)}. ¬ a !! i"
  (is "_ ⟹ ∃ i ∈ {?l..<?r}. _")
  for a :: word32
proof (rule ccontr, simp del:word_size len_word ucast_bintr)
  {
    fix i
    have "(ucast kern_prefix :: word32) !! i = (i < size kern_prefix)"
      using nth_kern_prefix[of i] nth_ucast[of kern_prefix i] by auto
    moreover assume "i + ?l < ?r ⟹ a !! (i + ?l)"
    ultimately have "(a >> ?l) !! i = (ucast kern_prefix :: word32) !! i"
      using nth_shiftr[of a ?l i] by fastforce
  }
  moreover assume "∀ i∈{?l..<?r}. a !! i"
  ultimately have "a >> ?l = ucast kern_prefix" unfolding word_eq_iff using nth_ucast by auto
  moreover have "unat (a >> ?l) = unat a div 2 ^ ?l" using shiftr_div_2n' by blast
  moreover have "unat (ucast kern_prefix :: word32) = unat kern_prefix"
    by (rule unat_ucast_upcast, simp)
  ultimately have "unat a div 2 ^ ?l = unat kern_prefix" by simp
  hence "unat a ≥ unat kern_prefix ∗ 2 ^ ?l" by simp
  hence "unat a ≥ unat prefix_bound" unfolding rpad_def by simp
  also assume "unat a < unat prefix_bound"
  finally show False ..
qed

definition "set_of_write_cap w ≡ let (a, n) = ⌊w⌋ in {a .. a + of_nat n}" for w :: write_capability

adhoc_overloading abs set_of_write_cap

definition "write_cap_sub A B ≡
  let (a_A, n_A) = ⌊A⌋ in let (a_B, n_B) = ⌊B⌋ in a_B ≤ a_A ∧ a_A + of_nat n_A ≤ a_B + of_nat n_B"
  for A B :: write_capability

adhoc_overloading sub write_cap_sub

lemma write_cap_sub_iff[iff]: "a ⊆_c b = (⌈a⌉ ⊆ ⌈b⌉)" for a b :: write_capability
  unfolding write_cap_sub_def set_of_write_cap_def
  by (auto split:prod.splits)

lemmas write_cap_subsets[intro] = cap_sub.intro[OF write_cap_sub_iff]

interpretation write_cap_sub: cap_sub set_of_write_cap write_cap_sub ..

definition "write_cap_rep w ≡ let (a, n) = ⌊w⌋ in (a, of_nat n :: word32)"

adhoc_overloading rep write_cap_rep

lemma write_cap_inj[simp]: "(⌊w₁⌋ :: word32 × word32) = ⌊w₂⌋ ⟹ w₁ = w₂"
  for w₁ w₂ :: write_capability
  unfolding write_cap_rep_def
  by (auto
      split:prod.splits iff:write_cap_rep'_inject[symmetric]
```

```
    intro!:word_of_nat_inj simp add:rpad_def)

lemmas write_cap_invertible[intro] = invertible.intro[OF injI, OF write_cap_inj]

interpretation write_cap_inv: invertible write_cap_rep ..

adhoc_overloading abs write_cap_inv.inv

lemma write_cap_prefix[dest]: "a ∈ ⌈w⌉ ⟹ ¬ limited_and prefix_bound a" for w :: write_capability
proof
  assume "a ∈ ⌈w⌉"
  hence "unat a < unat prefix_bound"
    unfolding set_of_write_cap_def
    apply (simp split:prod.splits)
    using write_cap_bound'[of w] word_less_nat_alt word_of_nat_less by fastforce
  then obtain n where "n∈{LENGTH(256 word) − size kern_prefix..<LENGTH(256 word)}" and "¬ a !!
n"
    using write_cap_high[of a] by auto
  moreover assume "limited_and prefix_bound a"
  ultimately show False
    unfolding limited_and_def word_eq_iff
    by (subst (asm) nth_prefix_bound, auto)
qed

lemma write_cap_safe[simp]: "a ∈ ⌈w⌉ ⟹ a ≠ ⌊a'⌋" for w :: write_capability and a' :: address
  by auto
```

3.5.3 Log capability

```
typedef log_capability = "{ws :: word32 list. length ws ≤ 4}"
  morphisms log_cap_rep' log_capability
  by (intro exI[of _ "[]"], simp)

adhoc_overloading rep log_cap_rep'

definition "set_of_log_cap l ≡ {xs . prefix ⌊l⌋ xs}" for l :: log_capability

adhoc_overloading abs set_of_log_cap

definition "log_cap_sub A B ≡ prefix ⌊B⌋ ⌊A⌋" for A B :: log_capability

adhoc_overloading sub log_cap_sub

lemma log_cap_sub_iff[iff]: "a ⊆c b = (⌈a⌉ ⊆ ⌈b⌉)" for a b :: log_capability
  unfolding log_cap_sub_def set_of_log_cap_def
  by force

lemmas log_cap_subsets[intro] = cap_sub.intro[OF log_cap_sub_iff]

interpretation log_cap_sub: cap_sub set_of_log_cap log_cap_sub ..
```

Proof that that the log capability subset is defined according to the specification.

```
lemma "a ⊆c b = (∀ i < length ⌊b⌋ . ⌊a⌋ ! i = ⌊b⌋ ! i ∧ i < length ⌊a⌋)"
  (is "_ = ?R") for a b :: log_capability
  unfolding log_cap_sub_def prefix_def
proof
  let ?L = "∃ zs. ⌊a⌋ = ⌊b⌋ @ zs"
  {
    assume ?L
    moreover hence "length ⌊b⌋ ≤ length ⌊a⌋" by auto
    ultimately show "?L ⟹ ?R"
```

```
    by (auto simp add:nth_append)
  next
    assume ?R
    moreover hence len:"length ⌊b⌋ ≤ length ⌊a⌋"
      using le_def by blast
    moreover from ‹?R› have "⌊a⌋ = take (length ⌊b⌋) ⌊a⌋ @ drop (length ⌊b⌋) ⌊a⌋ "
      by simp
    moreover from ‹?R› len have "take (length ⌊b⌋) ⌊a⌋ = ⌊b⌋"
      by (metis nth_take_lemma order_refl take_all)
    ultimately show "?R ⟹ ?L" by (intro exI[of _ "drop (length ⌊b⌋) ⌊a⌋"], arith)
  }
qed

definition "log_cap_rep l ≡ (of_nat (length ⌊l⌋) :: word32) # ⌊l⌋"

no_adhoc_overloading rep log_cap_rep'

adhoc_overloading rep log_cap_rep

lemma log_cap_rep_inj[simp]: "(⌊l1⌋ :: word32 list) = ⌊l2⌋ ⟹ l1 = l2" for l1 l2 :: log_capability
  unfolding log_cap_rep_def using log_cap_rep'_inject by auto

lemmas log_cap_rep_invertible[intro] = invertible.intro[OF injI, OF log_cap_rep_inj]

interpretation log_cap_inv: invertible log_cap_rep ..
```

4 Kernel state

```
type_synonym eth_addr = "160 word" — 20 bytes

typedef 'a capability_list = "{l :: 'a list. length l < 2 ^ 8 − 1}"
  morphisms cap_list_rep cap_list
  by (intro exI[of _ "[]"], simp)

adhoc_overloading rep cap_list_rep

record procedure =
  eth_addr   :: eth_addr
  call_caps  :: "prefixed_capability capability_list"
  reg_caps   :: "prefixed_capability capability_list"
  del_caps   :: "prefixed_capability capability_list"
  entry_cap  :: bool
  write_caps :: "write_capability capability_list"

lemmas alist_simps = size_alist_def alist.Alist_inverse alist.impl_of_inverse

declare alist_simps[simp]

typedef procedure_list = "{l :: (key, procedure) alist. size l < 2 ^ LENGTH(key)}"
  morphisms proc_list_rep proc_list
  by (intro exI[of _ "Alist []"], simp)

adhoc_overloading rep proc_list_rep

record kernel =
  kern_addr  :: eth_addr
  curr_proc  :: eth_addr
  entry_proc :: eth_addr
  procs      :: procedure_list
```

4.1 Abbreviations

Here we introduce some useful abbreviations that will simplify the expression of the kernel state properties.

Number of the procedures:

```
abbreviation "nprocs σ ≡ size ⌊procs σ⌋"
```

Set of procedure indexes:

```
abbreviation "proc_ids σ ≡ {0..<nprocs σ}"
```

Procedure by its key:

```
abbreviation "proc σ k ≡ the (DAList.lookup ⌊procs σ⌋ k)"
```

Index of procedure:

Maximum number of procedures registered in the kernel:

```
abbreviation "max_nprocs ≡ 2 ^ LENGTH(key) − 1 :: nat"
```

5 Call formats

```
primrec split :: "'a::len word list ⇒ 'b::len word list list" where
  "split []        = []" |
  "split (x # xs) = word_rsplit x # split xs"

lemma cat_split[simp]: "map word_rcat (split x) = x"
  unfolding split_def
  by (induct x, simp_all add:word_rcat_rsplit)

lemma split_inj[simp]: "split x = split y ⟹ x = y"
  by (frule arg_cong[where f="map word_rcat"]) (subst (asm) cat_split)+

definition "maybe_inv2_tf z f l ≡
  if ∃ n. takefill z n l ∈ range2 f
  then Some (the_inv2 f (takefill z (SOME n. takefill z n l ∈ range2 f) l))
  else None"

lemma takefill_implies_prefix:
  assumes "x = takefill u n y"
  obtains (Prefix) "prefix x y" | (Postfix) "prefix y x"
proof (cases "length x ≤ length y")
  case True
  with assms have "prefix x y" unfolding takefill_alt by (simp add: take_is_prefix)
  with that show ?thesis by simp
next
  case False
  with assms have "prefix y x" unfolding takefill_alt by simp
  with that show ?thesis by simp
qed

lemma takefill_prefix_inj:
  "⟦⋀ x y. ⟦P x; P y; prefix x y⟧ ⟹ x = y; P x; P y; x = takefill u n y⟧ ⟹ x = y"
  by (elim takefill_implies_prefix) auto

lemma exI2[intro]: "P x y ⟹ ∃ x y. P x y" by auto

lemma maybe_inv2_tf_inj:
  "⟦⋀ x₁ y₁ x₂ y₂. prefix (f x₁ y₁) (f x₂ y₂) ⟹ x₁ = x₂;
    ⋀ x y y'. length (f x y) = length (f x y')⟧ ⟹ maybe_inv2_tf z f (f x y) = Some x"
```

```
  unfolding maybe_inv2_tf_def range2_def the_inv2_def
  apply (auto split:if_splits)
   apply (subst some1_equality[rotated], erule exI2)
     apply (metis length_takefill takefill_implies_prefix)
  apply (smt length_takefill takefill_implies_prefix the_equality)
  by (meson takefill_same)

lemma maybe_inv2_tf_inj':
  "⟦⋀ x₁ y₁ x₂ y₂. prefix (f x₁ y₁) (f x₂ y₂) ⟹ x₁ = x₂;
   ⋀ x y y'. length (f x y) = length (f x y')⟧ ⟹
   maybe_inv2_tf z f v = Some x ⟹ ∃ y n. f x y = takefill z n v"
  unfolding maybe_inv2_tf_def range2_def the_inv2_def
  apply (simp split:if_splits)
  apply (subst (asm) some1_equality[rotated], erule exI2)
   apply (metis length_takefill nat_less_le not_less take_prefix take_takefill)
  by (smt prefix_order.eq_iff the1_equality)

datatype result =
    Success storage
  | Revert

abbreviation "SYSCALL_NOEXIST ≡ 0xaa"

abbreviation "SYSCALL_BADCAP ≡ 0x33"

definition "cap_type_opt_rep c ≡ case c of Some c ⇒ ⌊c⌋ | None ⇒ 0x00"
  for c :: "capability option"

adhoc_overloading rep cap_type_opt_rep

lemma cap_type_opt_rep_inj[intro]: "inj cap_type_opt_rep" unfolding cap_type_opt_rep_def inj_def
  by (auto split:option.split)

lemmas cap_type_opt_invertible[intro] = invertible.intro[OF cap_type_opt_rep_inj]

interpretation cap_type_opt_inv: invertible cap_type_opt_rep ..

adhoc_overloading abs cap_type_opt_inv.inv

definition call :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "call _ _ s ≡ (Success s, [])"

definition register :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "register _ _ s ≡ (Success s, [])"

definition delete :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "delete _ _ s ≡ (Success s, [])"

definition set_entry :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "set_entry _ _ s ≡ (Success s, [])"

definition write_addr :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "write_addr _ _ s ≡ (Success s, [])"

definition log :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "log _ _ s ≡ (Success s, [])"

definition external :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "external _ _ s ≡ (Success s, [])"
```

```
definition execute :: "byte list ⇒ storage ⇒ result × byte list" where
  "execute c s ≡ case takefill 0x00 2 c of ct # ci # c ⇒
    (case ⌈ct⌉ of
      None          ⇒ (Revert, [SYSCALL_NOEXIST])
    | Some None      ⇒ (Success s, [])
    | Some (Some ct) ⇒ (case ⌈ci⌉ of
       None          ⇒ (Revert, [SYSCALL_BADCAP]) — Capability index out of bounds
     | Some ci       ⇒ (case ct of
         Call        ⇒ call ci c s
       | Reg         ⇒ register ci c s
       | Del         ⇒ delete ci c s
       | Entry       ⇒ set_entry ci c s
       | Write       ⇒ write_addr ci c s
       | Log         ⇒ log ci c s
       | Gas         ⇒ external ci c s)))"


end
```