Formal specification of the Cap9 kernel

Mikhail Mandrykin Ilya Shchepetkov

June 21, 2019

Contents

1 Introduction

This is an Isabelle/HOL theory that describes and proves the correctness of the Cap9 kernel specification.

2 Preliminaries

```
theory Cap9
imports
  "HOL−Word.Word"
  "HOL−Library.Adhoc_Overloading"
  "HOL−Library.DAList"
  "HOL−Library.Rewrite"
  "Word_Lib/Word_Lemmas"
begin
```

2.1 Type class instantiations

Instantiate *len* type class to extract lengths from word types avoiding repeated explicit numeric specification of the length e.g. $LENGTH(byte)$ or $LENGTH('a :: len\ word)$ instead of *8* or $LENGTH('a)$, where $'a$ cannot be directly extracted from a type such as $'a\ word$.

```
instantiation word :: (len) len begin
definition len_word[simp]: "len_of (_ :: 'a::len word itself) = LENGTH('a)"
instance by (standard, simp)
end

lemma len_word': "LENGTH('a::len word) = LENGTH('a)" by (rule len_word)
```

Instantiate *size* type class for types of the form $'a\ itself$. This allows us to parametrize operations by word lengths using the dummy variables of type $'a\ word\ itself$. The operations cannot be directly parametrized by numbers as there is no lifting from term numbers to type numbers due to the lack of dependent types.

```
instantiation itself :: (len) size begin
definition size_itself where [simp, code]: "size (n::'a::len itself) = LENGTH('a)"
instance ..
end

declare unat_word_ariths[simp] word_size[simp] is_up_def[simp] wsst_TYs(1,2)[simp]
```

2.2 Word width

We introduce definition of the least number of bits to hold the current value of a word. This is needed because in our specification we often word with $UCAST('a \to 'b)$'ed values (right aligned subranges of bits), largely again due to the lack of dependent types (or true type-level functions), e.g. the it's hard to specify that the length of $a \bowtie b$ (where $\bowtie$ stands for concatenation) is the sum of the length of a and b, since length is a type parameter and there's no equivalent of sum on the type level. So we instead fix the length of $a \bowtie b$ to be the maximum possible one (say, 32 bytes) and then use conditions of the form $width\ a \leq s$ to specify that the actual "size" of a is s.

```
definition "width w ≡ LEAST n. unat w < 2 ^ n" for w :: "'a::len word"

lemma widthI[intro]: "⟦⋀ u. u < n ⟹ 2 ^ u ≤ unat w; unat w < 2 ^ n⟧ ⟹ width w = n"
  unfolding width_def Least_def
  using not_le
  apply (intro the_equality, blast)
  by (meson nat_less_le)

lemma width_wf: "∃! n. (∀ u < n. 2 ^ u ≤ unat w) ∧ unat w < 2 ^ n"
  (is "?Ex1 (unat w)")
proof (induction ("unat w"))
  case 0
  show "?Ex1 0" by (intro ex1I[of _ 0], auto)
next
  case (Suc x)
```

```
  then obtain n where x:"(∀u<n. 2 ^ u ≤ x) ∧ x < 2 ^ n " by auto
  show  "?Ex1 (Suc x)"
  proof (cases "Suc x < 2 ^ n")
    case True
    thus "?Ex1 (Suc x)"
      using x
      apply (intro ex1I[of _ "n"], auto)
      by (meson Suc_lessD leD linorder_neqE_nat)
  next
    case False
    thus "?Ex1 (Suc x)"
      using x
      apply (intro ex1I[of _ "Suc n"], auto simp add: less_Suc_eq)
      apply (intro antisym)
       apply (metis One_nat_def Suc_lessI Suc_n_not_le_n leI numeral_2_eq_2 power_increasing_iff)
      by (metis Suc_lessD le_antisym not_le not_less_eq_eq)
  qed
qed

lemma width_iff[iff]: "(width w = n) = ((∀ u < n. 2 ^ u ≤ unat w) ∧ unat w < 2 ^ n)"
  using width_wf widthI by metis

lemma width_le_size: "width x ≤ size x"
proof−
  {
    assume "size x < width x"
    hence "2 ^ size x ≤ unat x" using width_iff by metis
    hence "2 ^ size x ≤ uint x" unfolding unat_def by simp
  }
  thus ?thesis using uint_range_size[of x] by (force simp del:word_size)
qed

lemma width_le_size'[simp]: "size x ≤ n ⟹ width x ≤ n" by (insert width_le_size[of x], simp)

lemma nth_width_high[simp]: "width x ≤ i ⟹ ¬ x !! i"
proof (cases "i < size x")
  case False
  thus ?thesis by (simp add: test_bit_bin')
next
  case True
  hence "(x < 2 ^ i) = (unat x < 2 ^ i)"
    unfolding unat_def
    using word_2p_lem by fastforce
  moreover assume "width x ≤ i"
  then obtain n where "unat x < 2 ^ n" and "n ≤ i" using width_iff by metis
  hence "unat x < 2 ^ i"
    by (meson le_less_trans nat_power_less_imp_less not_less zero_less_numeral)
  ultimately show ?thesis using bang_is_le by force
qed

lemma width_zero[iff]: "(width x = 0) = (x = 0)"
proof
  show "width x = 0 ⟹ x = 0" using nth_width_high[of x] word_eq_iff[of x 0] nth_0 by (metis le0)
  show "x = 0 ⟹ width x = 0" by simp
qed

lemma width_zero'[simp]: "width 0 = 0" by simp

lemma width_one[simp]: "width 1 = 1" by simp
```

```
lemma high_zeros_less: "(∀ i ≥ u. ¬ x !! i) ⟹ unat x < 2 ^ u"
  (is "?high ⟹ _") for x :: "'a::len word"
proof–
  assume ?high
  have size:"size (mask u :: 'a word) = size x" by simp
  {
    fix i
    from ‹?high› have "(x AND mask u) !! i = x !! i"
      using nth_mask[of u i] size test_bit_size[of x i]
      by (subst word_ao_nth) (elim allE[of _ i], auto)
  }
  with ‹?high› have "x AND mask u = x" using word_eq_iff by blast
  thus ?thesis unfolding unat_def using mask_eq_iff by auto
qed

lemma nth_width_msb[simp]: "x ≠ 0 ⟹ x !! (width x − 1)"
proof (rule ccontr)
  fix x :: "'a word"
  assume "x ≠ 0"
  hence width:"width x > 0" using width_zero by fastforce
  assume "¬ x !! (width x − 1)"
  with width have "∀ i ≥ width x − 1. ¬ x !! i"
    using nth_width_high[of x] antisym_conv2 by fastforce
  hence "unat x < 2 ^ (width x − 1)" using high_zeros_less[of "width x − 1" x] by simp
  moreover from width have "unat x ≥ 2 ^ (width x − 1)" using width_iff[of x "width x"] by simp
  ultimately show False by simp
qed

lemma width_iff': "((∀ i > u. ¬ x !! i) ∧ x !! u) = (width x = Suc u)"
proof (rule; (elim conjE | intro conjI))
  assume "x !! u" and "∀ i > u. ¬ x !! i"
  show "width x = Suc u"
  proof (rule antisym)
    from ‹x !! u› show "width x ≥ Suc u" using not_less nth_width_high by force
    from ‹x !! u› have "x ≠ 0" by auto
    with ‹∀ i > u. ¬ x !! i› have "width x − 1 ≤ u" using not_less nth_width_msb by metis
    thus "width x ≤ Suc u" by simp
  qed
next
  assume "width x = Suc u"
  show "∀ i>u. ¬ x !! i" by (simp add:‹width x = Suc u›)
  from ‹width x = Suc u› show "x !! u" using nth_width_msb width_zero
    by (metis diff_Suc_1 old.nat.distinct(2))
qed

lemma width_word_log2: "x ≠ 0 ⟹ width x = Suc (word_log2 x)"
  using word_log2_nth_same word_log2_nth_not_set width_iff' test_bit_size
  by metis

lemma width_ucast[OF refl, simp]: "uc = ucast ⟹ is_up uc ⟹ width (uc x) = width x"
  by (metis uint_up_ucast unat_def width_def)

lemma width_ucast'[OF refl, simp]:
  "uc = ucast ⟹ width x ≤ size (uc x) ⟹ width (uc x) = width x"
proof–
  have "unat x < 2 ^ width x" unfolding width_def by (rule LeastI_ex, auto)
  moreover assume "width x ≤ size (uc x)"
  ultimately have "unat x < 2 ^ size (uc x)" by (simp add: less_le_trans)
  moreover assume "uc = ucast"
  ultimately have "unat x = unat (uc x)" by (metis unat_ucast mod_less word_size)
```

```
  thus ?thesis unfolding width_def by simp
qed

lemma width_lshift[simp]:
  "⟦x ≠ 0; n ≤ size x − width x⟧ ⟹ width (x << n) = width x + n"
  (is "⟦_; ?nbound⟧ ⟹ _")
proof−
  assume "x ≠ 0"
  hence 0:"width x = Suc (width x − 1)" using width_zero by (metis Suc_pred' neq0_conv)
  from ‹x ≠ 0› have 1:"width x > 0" by (auto intro:gr_zeroI)
  assume ?nbound
  {
    fix i
    from ‹?nbound› have "i ≥ size x ⟹ ¬ x !! (i − n)" by (auto simp add:le_diff_conv2)
    hence "(x << n) !! i = (n ≤ i ∧ x !! (i − n))" using nth_shiftl'[of x n i] by auto
  } note corr = this
  hence "∀ i > width x + n − 1. ¬ (x << n) !! i" by auto
  moreover from corr have "(x << n) !! (width x + n − 1)"
    using width_iff'[of "width x − 1" x] 1
    by auto
  ultimately have "width (x << n) = Suc (width x + n − 1)" using width_iff' by auto
  thus ?thesis using 0 by simp
qed

lemma width_lshift'[simp]: "n ≤ size x − width x ⟹ width (x << n) ≤ width x + n"
  using width_zero width_lshift shiftl_0 by (metis eq_iff le0)

lemma width_or[simp]: "width (x OR y) = max (width x) (width y)"
proof−
  {
    fix a b
    assume "width x = Suc a" and "width y = Suc b"
    hence "width (x OR y) = Suc (max a b)"
      using width_iff' word_ao_nth[of x y] max_less_iff_conj[of "a" "b"]
      by (metis (no_types) max_def)
  } note succs = this
  thus ?thesis
  proof (cases "width x = 0 ∨ width y = 0")
    case True
    thus ?thesis using width_zero word_log_esimps(3,9) by (metis max_0L max_0R)
  next
    case False
    with succs show ?thesis by (metis max_Suc_Suc not0_implies_Suc)
  qed
qed
```

2.3 Right zero-padding

Here's the first time we use *width*. If x is a value of size n right-aligned in a word of size $s = size\ x$ (note there's nowhere to keep the value n, since the size of x is some $s \geq n$, so we require it to be provided explicitly), then *rpad* n x will move the value x to the left. For the operation to be correct (no losing of significant higher bits) we need the precondition $width\ x \leq n$ in all the lemmas, hence the need for *width*.

```
definition rpad where "rpad n x ≡ x << size x − n"

lemma rpad_low[simp]: "⟦width x ≤ n; i < size x − n⟧ ⟹ ¬ (rpad n x) !! i"
  unfolding rpad_def by (simp add:nth_shiftl)

lemma rpad_high[simp]:
```

```
  "⟦width x ≤ n; n ≤ size x; size x − n ≤ i⟧ ⟹ (rpad n x) !! i = x !! (i + n − size x)"
  (is "⟦?xbound; ?nbound; i ≥ ?ibound⟧ ⟹ ?goal i")
proof−
  fix i
  assume ?xbound ?nbound and "i ≥ ?ibound"
  moreover from ‹?nbound› have "i + n − size x = i − ?ibound" by simp
  moreover from ‹?xbound› have "x !! (i + n − size x) ⟹ i < size x" by − (rule ccontr, simp)
  ultimately show "?goal i" unfolding rpad_def by (subst nth_shiftl', metis)
qed

lemma rpad_inj: "⟦width x ≤ n; width y ≤ n; n ≤ size x⟧ ⟹ rpad n x = rpad n y ⟹ x = y"
  (is "⟦?xbound; ?ybound; ?nbound; _⟧ ⟹ _")
  unfolding inj_def word_eq_iff
proof (intro allI impI)
  fix i
  let ?i' = "i + size x − n"
  assume ?xbound ?ybound ?nbound
  assume "∀j < LENGTH('a). rpad n x !! j = rpad n y !! j"
  hence "⋀j. rpad n x !! j = rpad n y !! j" using test_bit_bin by blast
  from this[of ?i'] and ‹?xbound› ‹?ybound› ‹?nbound› show "x !! i = y !! i" by simp
qed
```

2.4 Spanning concatenation

```
abbreviation ucastl ("'(ucast')_ _" [1000, 100] 100) where
  "(ucast)_l a ≡ ucast a :: 'b word" for l :: "'b::len0 itself"

notation (input) ucastl ("'(ucast')_ _" [1000, 100] 100)

definition pad_join :: "'a::len word ⇒ nat ⇒ 'c::len itself ⇒ 'b::len word ⇒ 'c word"
  ("_ _◊_ _" [60, 1000, 1000, 61] 60) where
  "x n◊l y ≡ rpad n (ucast x) OR ucast y"

notation (input) pad_join ("_ _◊_ _" [60, 1000, 1000, 61] 60)

lemma pad_join_high:
  "⟦width a ≤ n; n ≤ size l; width b ≤ size l − n; size l − n ≤ i⟧
   ⟹ (a n◊l b) !! i = a !! (i + n − size l)"
  unfolding pad_join_def
  using nth_ucast nth_width_high by fastforce

lemma pad_join_high'[simp]:
  "⟦width a ≤ n; n ≤ size l; width b ≤ size l − n⟧ ⟹ a !! i = (a n◊l b) !! (i + size l − n)"
  using pad_join_high[of a n l b "i + size l − n"] by simp

lemma pad_join_mid[simp]:
  "⟦width a ≤ n; n ≤ size l; width b ≤ size l − n; width b ≤ i; i < size l − n⟧
   ⟹ ¬ (a n◊l b) !! i"
  unfolding pad_join_def by auto

lemma pad_join_low[simp]:
  "⟦width a ≤ n; n ≤ size l; width b ≤ size l − n; i < width b⟧ ⟹ (a n◊l b) !! i = b !! i"
  unfolding pad_join_def by (auto simp add: nth_ucast)

lemma pad_join_inj:
  assumes eq:"a n◊l b = c n◊l d"
  assumes a:"width a ≤ n" and c:"width c ≤ n"
  assumes n: "n ≤ size l"
  assumes b:"width b ≤ size l − n"
  assumes d:"width d ≤ size l − n"
  shows   "a = c" and "b = d"
```

```
proof−
  from eq have eq':"⋀j. (a n◇l b) !! j = (c n◇l d) !! j"
    using test_bit_bin unfolding word_eq_iff by auto
  moreover from a n b
  have "⋀ i. a !! i = (a n◇l b) !! (i + size l − n)" by simp
  moreover from c n d
  have "⋀ i. c !! i = (c n◇l d) !! (i + size l − n)" by simp
  ultimately show "a = c" unfolding word_eq_iff by auto

  {
    fix i
    from a n b have "i < width b ⟹ b !! i = (a n◇l b) !! i" by simp
    moreover from c n d have "i < width d ⟹ d !! i = (c n◇l d) !! i" by simp
    moreover have "i ≥ width b ⟹ ¬ b !! i" and "i ≥ width d ⟹ ¬ d !! i" by auto
    ultimately have "b !! i = d !! i"
      using eq'[of i] b d
        pad_join_mid[of a n l b i, OF a n b]
        pad_join_mid[of c n l d i, OF c n d]
      by (meson leI less_le_trans)
  }
  thus "b = d" unfolding word_eq_iff by simp
qed

lemma pad_join_inj'[dest!]:
 "⟦a n◇l b = c n◇l d;
   width a ≤ n; width c ≤ n; n ≤ size l;
   width b ≤ size l − n;
   width d ≤ size l − n⟧ ⟹ a = c ∧ b = d"
  apply (rule conjI)
  subgoal by (frule (4) pad_join_inj(1))
  by (frule (4) pad_join_inj(2))

lemma pad_join_and[simp]:
  assumes "width x ≤ n" "n ≤ m" "width a ≤ m" "m ≤ size l" "width b ≤ size l − m"
  shows   "(a m◇l b) AND rpad n x = rpad m a AND rpad n x"
  unfolding word_eq_iff
proof ((subst word_ao_nth)+, intro allI impI)
  from assms have 0:"n ≤ size x" by simp
  from assms have 1:"m ≤ size a" by simp
  fix i
  assume "i < LENGTH('a)"
  from assms show "((a m◇l b) !! i ∧ rpad n x !! i) = (rpad m a !! i ∧ rpad n x !! i)"
    using rpad_low[of x n i, OF assms(1)] rpad_high[of x n i, OF assms(1) 0]
          rpad_low[of a m i, OF assms(3)] rpad_high[of a m i, OF assms(3) 1]
          pad_join_high[of a m l b i, OF assms(3,4,5)]
          size_itself_def[of l] word_size[of x] word_size[of a]
    by (metis add.commute add_lessD1 le_Suc_ex le_diff_conv not_le)
qed
```

2.5 Deal with partially undefined results

```
definition restrict :: "'a::len word ⇒ nat set ⇒ 'a word" (infixl "↑" 60) where
  "restrict x s ≡ BITS i. i ∈ s ∧ x !! i"

lemma nth_restrict[iff]: "(x ↑ s) !! n = (n ∈ s ∧ x !! n)"
  unfolding restrict_def
  by (simp add: bang_conj_lt test_bit.eq_norm)

lemma restrict_inj2:
  assumes eq:"f x1 y1 OR v1 ↑ s = f x2 y2 OR v2 ↑ s"
  assumes fi:"⋀ x y i. i ∈ s ⟹ ¬ f x y !! i"
```

```
  assumes inj:"⋀ x1 y1 x2 y2. f x1 y1 = f x2 y2 ⟹ x1 = x2 ∧ y1 = y2"
  shows   "x1 = x2 ∧ y1 = y2"
proof–
  from eq and fi have "f x1 y1 = f x2 y2" unfolding word_eq_iff by auto
  with inj show ?thesis .
qed

lemma restrict_ucast_inv[simp]:
  "⟦a = LENGTH('a); b = LENGTH('b)⟧ ⟹ (ucast x OR y ↾ {a..<b}) AND mask a = ucast x"
  for x :: "'a::len word" and y :: "'b::len word"
  unfolding word_eq_iff
  by (rewrite nth_ucast word_ao_nth nth_mask nth_restrict test_bit_bin)+ auto

lemmas restrict_inj_pad_join[dest] = restrict_inj2[of "λ x y. x _◊_ y"]
```

2.6 Plain concatenation

```
definition join :: "'a::len word ⇒ 'c::len itself ⇒ nat ⇒ 'b::len word ⇒ 'c word"
  ("_ _⋈_ _" [62,1000,1000,61] 61) where
  "(a ₗ⋈ₙ b) ≡ (ucast a << n) OR (ucast b)"

notation (input) join ("_ _⋈_ _" [62,1000,1000,61] 61)

lemma width_join:
  "⟦width a + n ≤ size l; width b ≤ n⟧ ⟹ width (a ₗ⋈ₙ b) ≤ width a + n"
  (is "⟦?abound; ?bbound⟧ ⟹ _")
proof–
  assume ?abound and ?bbound
  moreover hence "width b ≤ size l" by simp
  ultimately show ?thesis
    using width_lshift'[of n "(ucast)ₗ a"]
    unfolding join_def
    by simp
qed

lemma width_join'[simp]:
  "⟦width a + n ≤ size l; width b ≤ n; width a + n ≤ q⟧ ⟹ width (a ₗ⋈ₙ b) ≤ q"
  by (drule (1) width_join, simp)

lemma join_high[simp]:
  "⟦width a + n ≤ size l; width b ≤ n; width a + n ≤ i⟧ ⟹ ¬ (a ₗ⋈ₙ b) !! i"
  by (drule (1) width_join, simp)

lemma join_mid:
  "⟦width a + n ≤ size l; width b ≤ n; n ≤ i; i < width a + n⟧ ⟹ (a ₗ⋈ₙ b) !! i = a !! (i − n)"
  apply (subgoal_tac "i < size ((ucast)ₗ a) ∧ size ((ucast)ₗ a) = size l")
  unfolding join_def
  using word_ao_nth nth_ucast nth_width_high nth_shiftl'
   apply (metis less_imp_diff_less order_trans word_size)
  by simp

lemma join_mid'[simp]:
  "⟦width a + n ≤ size l; width b ≤ n⟧ ⟹ a !! i = (a ₗ⋈ₙ b) !! (i + n)"
  using join_mid[of a n l b "i + n"] nth_width_high[of a i] join_high[of a n l b "i + n"]
  by force

lemma join_low[simp]:
  "⟦width a + n ≤ size l; width b ≤ n; i < n⟧ ⟹ (a ₗ⋈ₙ b) !! i = b !! i"
  unfolding join_def
  by (simp add: nth_shiftl nth_ucast)
```

```
lemma join_inj:
  assumes eq:"a ⌊⋈⌋_n b = c ⌊⋈⌋_n d"
  assumes "width a + n ≤ size l" and "width b ≤ n"
  assumes "width c + n ≤ size l" and "width d ≤ n"
  shows   "a = c" and "b = d"
proof–
  from assms show "a = c" unfolding word_eq_iff using join_mid' eq by metis
  from assms show "b = d" unfolding word_eq_iff using join_low nth_width_high
    by (metis eq less_le_trans not_le)
qed

lemma join_inj'[dest!]:
  "⟦a ⌊⋈⌋_n b = c ⌊⋈⌋_n d;
    width a + n ≤ size l; width b ≤ n;
    width c + n ≤ size l; width d ≤ n⟧ ⟹ a = c ∧ b = d"
  apply (rule conjI)
  subgoal by (frule (4) join_inj(1))
  by (frule (4) join_inj(2))

lemma join_and:
  assumes "width x ≤ n" "n ≤ size l" "k ≤ size l" "m ≤ k"
          "n ≤ k − m" "width a ≤ k − m" "width a + m ≤ k" "width b ≤ m"
  shows   "rpad k (a ⌊⋈⌋_m b) AND rpad n x = rpad (k − m) a AND rpad n x"
  unfolding word_eq_iff
proof ((subst word_ao_nth)+, intro allI impI)
  from assms have 0:"n ≤ size x" by simp
  from assms have 1:"k − m ≤ size a" by simp
  from assms have 2:"width (a ⌊⋈⌋_m b) ≤ k" by simp
  from assms have 3:"k ≤ size (a ⌊⋈⌋_m b)" by simp
  from assms have 4:"width a + m ≤ size l" by simp
  fix i
  assume "i < LENGTH('a)"
  moreover with assms have "i + k − size (a ⌊⋈⌋_m b) − m = i + (k − m) − size a" by simp
  moreover from assms have "i + k − size (a ⌊⋈⌋_m b) < m ⟹ i < size x − n" by simp
  moreover from assms have
    "⟦i ≥ size l − k; m ≤ i + k − size (a ⌊⋈⌋_m b)⟧ ⟹ size a − (k − m) ≤ i" by simp
  moreover from assms have "width a + m ≤ i + k − size (a ⌊⋈⌋_m b) ⟹ ¬ rpad (k − m) a !! i"
    by (simp add: nth_shiftl' rpad_def)
  moreover from assms have "¬ i ≥ size l − k ⟹ i < size x − n" by simp
  ultimately show "(rpad k (a ⌊⋈⌋_m b) !! i ∧ rpad n x !! i) =
                   (rpad (k − m) a !! i ∧ rpad n x !! i)"
    using assms
          rpad_high[of x n i, OF assms(1) 0] rpad_low[of x n i, OF assms(1)]
          rpad_high[of a "k − m" i, OF assms(6) 1] rpad_low[of a "k − m" i, OF assms(6)]
          rpad_high[of "a ⌊⋈⌋_m b" k i, OF 2 3] rpad_low[of "a ⌊⋈⌋_m b" k i, OF 2]
          join_high[of a m l b "i + k − size (a ⌊⋈⌋_m b)", OF 4 assms(8)]
          join_mid[of a m l b "i + k − size (a ⌊⋈⌋_m b)", OF 4 assms(8)]
          join_low[of a m l b "i + k − size (a ⌊⋈⌋_m b)", OF 4 assms(8)]
          size_itself_def[of l] word_size[of x] word_size[of a] word_size[of "a ⌊⋈⌋_m b"]
    by (metis not_le)
qed

lemma join_and'[simp]:
  "⟦width x ≤ n; n ≤ size l; k ≤ size l; m ≤ k;
    n ≤ k − m; width a ≤ k − m; width a + m ≤ k; width b ≤ m⟧ ⟹
   rpad k (a ⌊⋈⌋_m b) AND rpad n x = rpad (k − m) (ucast a) AND rpad n x"
  using join_and[of x n l k m "ucast a" b] unfolding join_def
  by (simp add: ucast_id)
```

3 Data formats

This section contains definitions of various data formats used in the specification.

3.1 Common notation

Before we proceed some common notation that would be used later will be established.

3.1.1 Machine words

Procedure keys are represented as 24-byte (192 bits) machine words.

type_synonym *word24* = *"192 word"* — 24 bytes
type_synonym *key* = *word24*

Byte is 8-bit machine word.

type_synonym *byte* = *"8 word"*

32-byte machine words that are used to model keys and values of the storage.

type_synonym *word32* = *"256 word"* — 32 bytes

Storage is a function that takes a 32-byte word (key) and returns another 32-byte word (value).

type_synonym *storage* = *"word32 ⇒ word32"*

3.1.2 Concatenation operations

Specialize previously defined general concatenation operations for the fixed result size of 32 bytes. Thus we avoid lots of redundant type annotations for every intermediate result (note that these intermediate types cannot be inferred automatically (in a purely Hindley-Milner setting as in Isabelle), because this would require type-level functions/dependent types).

abbreviation *"len (_ :: 'a::len word itself) ≡ TYPE('a)"*

no_notation *join* ("_ _⋈_ _" [62,1000,1000,61] 61)
no_notation (*input*) *join* ("_ _⋈_ _" [62,1000,1000,61] 61)

abbreviation *join32* ("_ ⋈_ _" [62,1000,61] 61) **where**
  *"a ⋈n b ≡ join a (len TYPE(word32)) (n * 8) b"*
abbreviation (**output**) *join32_out* ("_ ⋈_ _" [62,1000,61] 61) **where**
  "join32_out a n b ≡ join a (TYPE(256)) n b"
notation (*input*) *join32* ("_ ⋈_ _" [62,1000,61] 61)

no_notation *pad_join* ("_ _◊_ _" [60,1000,1000,61] 60)
no_notation (*input*) *pad_join* ("_ _◊_ _" [60,1000,1000,61] 60)

abbreviation *pad_join32* ("_ _◊ _" [60,1000,61] 60) **where**
  *"a n◊ b ≡ pad_join a (n * 8) (len TYPE(word32)) b"*
abbreviation (**output**) *pad_join32_out* ("_ _◊ _" [60,1000,61] 60) **where**
  "pad_join32_out a n b ≡ pad_join a n (TYPE(256)) b"
notation (*input*) *pad_join32* ("_ _◊ _" [60,1000,61] 60)

Override treatment of hexidecimal numeric constants to make them monomorphic words of fixed length, mimicking the notation used in the informal specification (e.g. $1::'a$) is always a word 1 byte long and is not, say, the natural number one). Otherwise, again, lots of redundant type annotations would arise.

parse_ast_translation ‹
  let
    open Ast

```
  fun mk_numeral t = mk_appl (Constant @{syntax_const _Numeral}) t
  fun mk_word_numeral num t =
    if String.isPrefix 0x num then
      mk_appl (Constant @{syntax_const _constrain})
        [mk_numeral t,
         mk_appl (Constant @{type_syntax word})
           [mk_appl (Constant @{syntax_const _NumeralType})
           [Variable (4 * (size num - 2) |> string_of_int)]]]
    else
      mk_numeral t
  fun numeral_ast_tr ctxt (t as [Appl [Constant @{syntax_const _constrain},
                                       Constant num,
                                       _]])
                                           = mk_word_numeral num t
    | numeral_ast_tr ctxt (t as [Constant num]) = mk_word_numeral num t
    | numeral_ast_tr _ t                       = mk_numeral t
    | numeral_ast_tr _ t                       = raise AST (@{syntax_const _Numeral}, t)
in
  [(@{syntax_const _Numeral}, numeral_ast_tr)]
end
›
```

3.2 Datatypes

Introduce generic notation for mapping of various entities into high-level and low-level representations. A high-level representation of an entity e would be written as $\lceil e \rceil$ and a low-level as $\lfloor e \rfloor$ accordingly. Using a high-level representation it is easier to express and proof some properties and invariants, but some of them can be expressed only using a low-level representation.

We use adhoc overloading to use the same notation for various types of entities (indices, offsets, addresses, capabilities etc.).

```
no_notation floor ("⌊_⌋")

consts rep :: "'a ⇒ 'b" ("⌊_⌋")

no_notation ceiling ("⌈_⌉")

consts abs :: "'a ⇒ 'b" ("⌈_⌉")
```

3.2.1 Deterministic inverse functions

```
definition "maybe_inv f y ≡ if y ∈ range f then Some (the_inv f y) else None"

lemma maybe_inv_inj[intro]: "inj f ⟹ maybe_inv f (f x) = Some x"
  unfolding maybe_inv_def
  by (auto simp add:inj_def the_inv_f_f)

lemma maybe_inv_inj'[dest]: "⟦inj f; maybe_inv f y = Some x⟧ ⟹ f x = y"
  unfolding maybe_inv_def
  by (auto intro:f_the_inv_into_f simp add:inj_def split:if_splits)

locale invertible =
  fixes rep :: "'a ⇒ 'b" ("⌊_⌋")
  assumes inj:"inj rep"
begin
definition inv :: "'b ⇒ 'a option" where "inv ≡ maybe_inv rep"

lemmas inv_inj[folded inv_def, simp] = maybe_inv_inj[OF inj]

lemmas inv_inj'[folded inv_def, dest] = maybe_inv_inj'[OF inj]
```

```
end

definition "range2 f ≡ {y. ∃x₁ ∈ UNIV. ∃ x₂ ∈ UNIV. y = f x₁ x₂}"

definition "the_inv2 f ≡ λ x. THE y. ∃ y'. f y y' = x"

definition "maybe_inv2 f y ≡ if y ∈ range2 f then Some (the_inv2 f y) else None"

definition "inj2 f ≡ ∀ x₁ x₂ y₁ y₂. f x₁ y₁ = f x₂ y₂ ⟶ x₁ = x₂"

lemma inj2I: "(⋀ x₁ x₂ y₁ y₂. f x₁ y₁ = f x₂ y₂ ⟹ x₁ = x₂) ⟹ inj2 f" unfolding inj2_def
  by blast

lemma maybe_inv2_inj[intro]: "inj2 f ⟹ maybe_inv2 f (f x y) = Some x"
  unfolding maybe_inv2_def the_inv2_def inj2_def range2_def
  by (simp split:if_splits, blast)

lemma maybe_inv2_inj'[dest]:
  "⟦inj2 f; maybe_inv2 f y = Some x⟧ ⟹ ∃ y'. f x y' = y"
  unfolding maybe_inv2_def the_inv2_def range2_def inj2_def
  by (force split:if_splits intro:theI)

locale invertible2 =
  fixes rep :: "'a ⇒ 'b ⇒ 'c" ("⌊_⌋")
  assumes inj:"inj2 rep"
begin
definition inv2 :: "'c ⇒ 'a option" where "inv2 ≡ maybe_inv2 rep"

lemmas inv2_inj[folded inv2_def, simp] = maybe_inv2_inj[OF inj]

lemmas inv2_inj'[folded inv_def, dest] = maybe_inv2_inj'[OF inj]
end
```

3.2.2 Capability

Introduce capability type. Note that we don't include *Null* capability into it. *Null* is only handled specially inside the call delegation, otherwise it only complicates the proofs with side additional cases.

There will be separate type *call* defined as *capability option* to respect the fact that in some places it can indeed be *Null*.

```
datatype capability =
    Call
  | Reg
  | Del
  | Entry
  | Write
  | Log
  | Send
```

In general, in the following we strive to make all encoding functions injective without any preconditions. All the necessary invariants are built into the type definitions.

Capability representation would be its assigned number.

```
definition cap_type_rep :: "capability ⇒ byte" where
  "cap_type_rep c ≡ case c of
     Call  ⇒ 0x03
   | Reg   ⇒ 0x04
   | Del   ⇒ 0x05
   | Entry ⇒ 0x06
   | Write ⇒ 0x07
   | Log   ⇒ 0x08
```

```
  | Send ⇒ 0x09"
```

```
adhoc_overloading rep cap_type_rep
```

Capability representation range from *3* to *9* since *Null* is not included and *2* does not exist.

```
lemma cap_type_rep_rng[simp]: "⌊c⌋ ∈ {0x03..0x09}" for c :: capability
  unfolding cap_type_rep_def by (simp split:capability.split)
```

Capability representation is injective.

```
lemma cap_type_rep_inj[dest]: "⌊c1⌋ = ⌊c2⌋ ⟹ c1 = c2" for c1 c2 :: capability
  unfolding cap_type_rep_def
  by (simp split:capability.splits)
```

4 bits is sufficient to store a capability number.

```
lemma width_cap_type: "width ⌊c⌋ ≤ 4" for c :: capability
proof (rule ccontr, drule not_le_imp_less)
  assume "4 < width ⌊c⌋"
  moreover hence "⌊c⌋ !! (width ⌊c⌋ − 1)" using nth_width_msb by force
  ultimately obtain n where "⌊c⌋ !! n" and "n ≥ 4" by (metis le_step_down_nat nat_less_le)
  thus False unfolding cap_type_rep_def by (simp split:capability.splits)
qed
```

So, any number greater than or equal to *4* will be enough.

```
lemma width_cap_type'[simp]: "4 ≤ n ⟹ width ⌊c⌋ ≤ n" for c :: capability
  using width_cap_type[of c] by simp
```

Capability representation can't be zero.

```
lemma cap_type_nonzero[simp]: "⌊c⌋ ≠ 0" for c :: capability
  unfolding cap_type_rep_def by (simp split:capability.splits)
```

3.2.3 Capability index

Introduce capability index type that is a natural number in range from 0 to 254.

```
typedef capability_index = "{i :: nat. i < 2 ^ LENGTH(byte) − 1}"
  morphisms cap_index_rep' cap_index
  by (intro exI[of _ "0"], simp)
```

```
adhoc_overloading rep cap_index_rep'
```

```
adhoc_overloading abs cap_index
```

Capability index representation is a byte. Zero byte is reserved, so capability index representation starts with 1.

```
definition "cap_index_rep i ≡ of_nat (⌊i⌋ + 1) :: byte" for i :: capability_index
```

```
adhoc_overloading rep cap_index_rep
```

A single byte is sufficient to store the least number of bits of capability index representation.

```
lemma width_cap_index: "width ⌊i⌋ ≤ LENGTH(byte)" for i :: capability_index by simp
```

```
lemma width_cap_index'[simp]: "LENGTH(byte) ≤ n ⟹ width ⌊i⌋ ≤ n"
  for i :: capability_index by simp
```

Capability index representation can't be zero byte.

```
lemma cap_index_nonzero[simp]: "⌊i⌋ ≠ 0x00" for i :: capability_index
  unfolding cap_index_rep_def using cap_index_rep'[of i] of_nat_neq_0[of "Suc ⌊i⌋"]
  by force
```

Capability index representation is injective.

```
lemma cap_index_inj[dest]: "(⌊i₁⌋ :: byte) = ⌊i₂⌋ ⟹ i₁ = i₂" for i₁ i₂ :: capability_index
  unfolding cap_index_rep_def
  using cap_index_rep'[of i₁] cap_index_rep'[of i₂] word_of_nat_inj[of "⌊i₁⌋" "⌊i₂⌋"]
        cap_index_rep'_inject
  by force
```

Representation function is invertible.

```
lemmas cap_index_invertible[intro] = invertible.intro[OF injI, OF cap_index_inj]

interpretation cap_index_inv: invertible cap_index_rep ..

adhoc_overloading abs cap_index_inv.inv
```

3.2.4 Capability offset

The following datatype specifies data offsets for addresses in the procedure heap.

```
type_synonym capability_offset = byte

datatype data_offset =
  Addr
  | Index
  | Ncaps capability
  | Cap capability capability_index capability_offset
```

Machine word representation of data offsets. Using these offsets the following data can be obtained:

- *Addr*: procedure Ethereum address;
- *Index*: procedure index;
- *Ncaps ty*: the number of capabilities of type *ty*;
- *Cap ty i off*: capability of type *ty*, with index *ty* and offset *off* into that capability.

```
definition data_offset_rep :: "data_offset ⇒ word32" where
 "data_offset_rep off ≡ case off of
    Addr       ⇒ 0x00 ⋈₂ 0x00 ⋈₁ 0x00
  | Index      ⇒ 0x00 ⋈₂ 0x00 ⋈₁ 0x01
  | Ncaps ty   ⇒ ⌊ty⌋ ⋈₂ 0x00 ⋈₁ 0x00
  | Cap ty i off ⇒ ⌊ty⌋ ⋈₂ ⌊i⌋ ⋈₁ off"

adhoc_overloading rep data_offset_rep
```

Data offset representation is injective.

```
lemma data_offset_inj[dest]:
  "⌊d₁⌋ = ⌊d₂⌋ ⟹ d₁ = d₂" for d₁ d₂ :: data_offset
  unfolding data_offset_rep_def
  by (auto split:data_offset.splits)
```

Least number of bytes to hold the current value of a data offset is *3*.

```
lemma width_data_offset: "width ⌊d⌋ ≤ 3 * LENGTH(byte)" for d :: data_offset
  unfolding data_offset_rep_def
  by (simp split:data_offset.splits)

lemma width_data_offset'[simp]: "3 * LENGTH(byte) ≤ n ⟹ width ⌊d⌋ ≤ n" for d :: data_offset
  using width_data_offset[of d] by simp
```

3.2.5 Kernel storage address

Type definition for procedure indices. A procedure index is represented as a natural number that is smaller then $2^{192} - 1$. It can be zero here, to simplify its future use as an array index, but its low-level representation will start from *1*.

```
typedef key_index = "{i :: nat. i < 2 ^ LENGTH(key) - 1}" morphisms key_index_rep' key_index
  by (rule exI[of _ "0"], simp)

adhoc_overloading rep key_index_rep'

adhoc_overloading abs key_index
```

Introduce address datatype that describes possible addresses in the kernel storage.

```
datatype address =
   Heap_proc key data_offset
 | Nprocs
 | Proc_key key_index
 | Kernel
 | Curr_proc
 | Entry_proc
```

Low-level representation of a procedure index is a machine word that starts from *1*.

```
definition "key_index_rep i ≡ of_nat (⌊i⌋ + 1) :: key" for i :: key_index

adhoc_overloading rep key_index_rep
```

Proof that low-level representation can't be *0*.

```
lemma key_index_nonzero[simp]: "⌊i⌋ ≠ (0 :: key)" for i :: key_index
  unfolding key_index_rep_def using key_index_rep'[of i]
  by (intro of_nat_neq_0, simp_all)
```

Low-level representation is injective.

```
lemma key_index_inj[dest]: "(⌊i₁⌋ :: key) = ⌊i₂⌋ ⟹ i₁ = i₂" for i :: key_index
  unfolding key_index_rep_def using key_index_rep'[of i₁] key_index_rep'[of i₂]
  by (simp add:key_index_rep'_inject of_nat_inj)
```

Address prefix for all addresses that belong to the kernel storage.

```
abbreviation "kern_prefix ≡ 0xffffffff"
```

Machine word representation of the kernel storage layout, which consists of the following addresses:

- *Heap_proc k offs*: procedure heap of key k and data offset *offs*;
- *Nprocs*: number of procedures;
- *Proc_key i*: a procedure with index i in the procedure list;
- *Kernel*: kernel Ethereum address;
- *Curr_proc*: current procedure;
- *Entry_proc*: entry procedure.

```
definition addr_rep :: "address ⇒ word32" where
  "addr_rep a ≡ case a of
    Heap_proc k offs ⇒ kern_prefix ⋈₁ 0x00 ₅◊ k          ⋈₃ ⌊offs⌋
  | Nprocs           ⇒ kern_prefix ⋈₁ 0x01 ₅◊ (0 :: key) ⋈₃ 0x000000
  | Proc_key i       ⇒ kern_prefix ⋈₁ 0x01 ₅◊ ⌊i⌋        ⋈₃ 0x000000
  | Kernel           ⇒ kern_prefix ⋈₁ 0x02 ₅◊ (0 :: key) ⋈₃ 0x000000
```

```
  | Curr_proc      ⇒ kern_prefix ⋈₁ 0x03 ₅◊ (0 :: key) ⋈₃ 0x000000
  | Entry_proc    ⇒ kern_prefix ⋈₁ 0x04 ₅◊ (0 :: key) ⋈₃ 0x000000"

adhoc_overloading rep addr_rep
```

Kernel storage address representation is injective.

```
lemma addr_inj[dest]: "⌊a₁⌋ = ⌊a₂⌋ ⟹ a₁ = a₂" for a₁ a₂ :: address
  unfolding addr_rep_def
  by (split address.splits) (force split:address.splits)+
```

Representation function is invertible.

```
lemmas addr_invertible[intro] = invertible.intro[OF injI, OF addr_inj]

interpretation addr_inv: invertible addr_rep ..

adhoc_overloading abs addr_inv.inv
```

Lowest address of the kernel storage (0xffffffff0000...).

```
abbreviation "prefix_bound ≡ rpad (size kern_prefix) (ucast kern_prefix :: word32)"

lemma prefix_bound: "unat prefix_bound < 2 ^ LENGTH(word32)" unfolding rpad_def by simp

lemma prefix_bound'[simplified, simp]: "x ≤ unat prefix_bound ⟹ x < 2 ^ LENGTH(word32)"
  using prefix_bound by simp
```

All addresses in the kernel storage are indeed start with the kernel prefix (0xffffffff).

```
lemma addr_prefix[simp, intro]: "limited_and prefix_bound ⌊a⌋" for a :: address
  unfolding limited_and_def addr_rep_def
  by (subst word_bw_comms) (auto split:address.split simp del:ucast_bintr)
```

3.3 Capability formats

We define capability format generally as a *locale*. It has two parameters: first one is a *subset* function (denoted as $\subseteq_c$), and second one is a *set_of* function, which maps a capability to its high-level representation that is expressed as a set. We have an assumption that *Capability A* is a subset of *Capability B* if and only if their high-level representations are also subsets of each other. We call it the well-definedness assumption (denoted as wd) and using it we can prove abstractly that such generic capability format satisfies the properties of reflexivity and transitivity.

Then using this locale we can prove that capability formats of all available system calls satisfy the properties of reflexivity and transitivity simply by formalizing corresponding *subset* and *set_of* functions and then proving the well-definedness assumption. This process is called locale interpretation.

```
no_notation abs ("⌈_⌉")

locale cap_sub =
  fixes set_of :: "'a ⇒ 'b set" ("⌈_⌉")
  fixes sub :: "'a ⇒ 'a ⇒ bool" ("(_/ ⊆c _)" [51, 51] 50)
  assumes wd:"a ⊆c b = (⌈a⌉ ⊆ ⌈b⌉)" begin

lemma sub_refl: "a ⊆c a" using wd by auto

lemma sub_trans: "⟦a ⊆c b; b ⊆c c⟧ ⟹ a ⊆c c" using wd by blast
end

notation abs ("⌈_⌉")

consts sub :: "'a ⇒ 'a ⇒ bool" ("(_/ ⊆c _)" [51, 51] 50)
```

3.3.1 Call, Register and Delete capabilities

Call, Register and Delete capabilities have the same format, so we combine them together here. The capability format defines a range of procedure keys that the capability allows one to call. This is defined as a base procedure key and a prefix.

Prefix is defined as a natural number, whose length is bounded by a maximum length of a procedure key.

```
typedef prefix_size = "{n :: nat. n ≤ LENGTH(key)}"
  morphisms prefix_size_rep' prefix_size
  by auto

adhoc_overloading rep prefix_size_rep'
```

Low-level representation of a prefix is a 8-bit machine word (or simply a byte).

```
definition "prefix_size_rep s ≡ of_nat ⌊s⌋ :: byte" for s :: prefix_size

adhoc_overloading rep prefix_size_rep
```

Prefix representation is injective.

```
lemma prefix_size_inj[dest]: "(⌊s1⌋ :: byte) = ⌊s2⌋ ⟹ s1 = s2" for s1 s2 :: prefix_size
  unfolding prefix_size_rep_def using prefix_size_rep'[of s1] prefix_size_rep'[of s2]
  by (simp add:prefix_size_rep'_inject of_nat_inj)
```

Any number that is greater or equal to a maximum length of a procedure key is greater or equal to any procedure index.

```
lemma prefix_size_rep_less[simp]: "LENGTH(key) ≤ n ⟹ ⌊s⌋ ≤ (n :: nat)" for s :: prefix_size
  using prefix_size_rep'[of s] by simp
```

Capabilities that have the same format based on prefixes we call "prefixed". Type of prefixed capabilities is defined as a direct product of prefixes and procedure keys.

```
type_synonym prefixed_capability = "prefix_size × key"
```

High-level representation of a prefixed capability is a set of all procedure keys whose first s number of bits (specified by the prefix) are the same as the first s number of bits of the base procedure key k.

```
definition
  "set_of_pref_cap sk ≡ let (s, k) = sk in {k' :: key. take ⌊s⌋ (to_bl k') = take ⌊s⌋ (to_bl k)}"
  for sk :: prefixed_capability

adhoc_overloading abs set_of_pref_cap
```

A prefixed capability A is a subset of a prefixed capability B if:

- the prefix size of A is equal to or greater than the prefix size of B;
- the first s bits (specified by the prefix size of B) of the base procedure of A is equal to the first s bits of the base procedure of B.

```
definition "pref_cap_sub A B ≡
  let (sA, kA) = A; (sB, kB) = B in
  (⌊sA⌋ :: nat) ≥ ⌊sB⌋ ∧ take ⌊sB⌋ (to_bl kA) = take ⌊sB⌋ (to_bl kB)"
  for A B :: prefixed_capability

adhoc_overloading sub pref_cap_sub
```

Auxiliary lemma: if first n elements of lists a and b are equal, and the number i is smaller than n, then the ith elements of both lists are also equal.

```
lemma nth_take_i[dest]: "⟦take n a = take n b; i < n⟧ ⟹ a ! i = b ! i"
```

```
  by (metis nth_take)

lemma take_less_diff:
  fixes l' l'' :: "'a list"
  assumes ex:"⋀ u :: 'a. ∃ u'. u' ≠ u"
  assumes "n < m"
  assumes "length l' = length l''"
  assumes "n ≤ length l'"
  assumes "m ≤ length l'"
  obtains l where
      "length l = length l'"
  and "take n l = take n l'"
  and "take m l ≠ take m l''"
proof−
  let ?x = "l'' ! n"
  from ex obtain y where neq:"y ≠ ?x" by auto
  let ?l = "take n l' @ y # drop (n + 1) l'"
  from assms have 0:"n = length (take n l') + 0" by simp
  from assms have "take n ?l = take n l'" by simp
  moreover from assms and neq have "take m ?l ≠ take m l''"
    using 0 nth_take_i nth_append_length
    by (metis add.right_neutral)
  moreover have "length ?l = length l'" using assms by auto
  ultimately show ?thesis using that by blast
qed
```

Prove the well-definedness assumption for the prefixed capability format.

```
lemma pref_cap_sub_iff[iff]: "a ⊆c b = (⌈a⌉ ⊆ ⌈b⌉)" for a b :: prefixed_capability
proof
  show "a ⊆c b ⟹ ⌈a⌉ ⊆ ⌈b⌉"
    unfolding pref_cap_sub_def set_of_pref_cap_def
    by (force intro:nth_take_lemma)
  {
    fix n m :: prefix_size
    fix x y :: key
    assume "⌊n⌋ < (⌊m⌋ :: nat)"
    then obtain z where
      "length z = size x"
      "take ⌊n⌋ z = take ⌊n⌋ (to_bl x)" and "take ⌊m⌋ z ≠ take ⌊m⌋ (to_bl y)"
      using take_less_diff[of "⌊n⌋" "⌊m⌋" "to_bl x" "to_bl y"]
      by auto
    moreover hence "to_bl (of_bl z :: key) = z" by (intro word_bl.Abs_inverse[of z], simp)
    ultimately
    have "∃ u :: key.
          take ⌊n⌋ (to_bl u) = take ⌊n⌋ (to_bl x) ∧ take ⌊m⌋ (to_bl u) ≠ take ⌊m⌋ (to_bl y)"
      by metis
  }
  thus "⌈a⌉ ⊆ ⌈b⌉ ⟹ a ⊆c b"
    unfolding pref_cap_sub_def set_of_pref_cap_def subset_eq
    apply (auto split:prod.split)
    by (erule contrapos_pp[of "∀ x. _ x"], simp)
qed

lemmas pref_cap_subsets[intro] = cap_sub.intro[OF pref_cap_sub_iff]
```

Locale interpretation to prove the reflexivity and transitivity properties of a subset function of the prefixed capability format.

```
interpretation pref_cap_sub: cap_sub set_of_pref_cap pref_cap_sub ..
```

Low-level 32-byte machine word representation of the prefixed capability format:

- first byte is the prefix;
- next seven bytes are undefined;
- 24 bytes of the base procedure key.

```
definition "pref_cap_rep sk r ≡
  let (s, k) = sk in ⌊s⌋ ₁◊ k OR r ↑ {LENGTH(key)..<LENGTH(word32) − LENGTH(byte)}"
  for sk :: prefixed_capability

adhoc_overloading rep pref_cap_rep
```

Low-level representation is injective.

```
lemma pref_cap_rep_inj_helper_inj[dest]: "⌊s₁⌋ ₁◊ k₁ = ⌊s₂⌋ ₁◊ k₂ ⟹ s₁ = s₂ ∧ k₁ = k₂"
  for s₁ s₂ :: prefix_size and k₁ k₂ :: key
  by auto

lemma pref_cap_rep_inj_helper_zero[simplified, simp]:
  "n ∈ {LENGTH(key)..<LENGTH(word32) − LENGTH(byte)} ⟹ ¬ (⌊s⌋ ₁◊ k) !! n"
  for s :: prefix_size and k :: key
  by simp

lemma pref_cap_rep_inj[dest]: "⌊c₁⌋ r₁ = ⌊c₂⌋ r₂ ⟹ c₁ = c₂" for c₁ c₂ :: prefixed_capability
  unfolding pref_cap_rep_def
  by (auto split:prod.splits)
```

Representation function is invertible.

```
lemmas pref_cap_invertible[intro] = invertible2.intro[OF inj2I, OF pref_cap_rep_inj]

interpretation pref_cap_inv: invertible2 pref_cap_rep ..

adhoc_overloading abs pref_cap_inv.inv2
```

3.3.2 Write capability

The write capability format includes 2 values: the first is the base address where we can write to storage. The second is the number of additional addresses we can write to.

Note that write capability must not allow to write to the kernel storage.

```
typedef write_capability = "{(a :: word32, n). n < unat prefix_bound − unat a}"
  morphisms write_cap_rep' write_cap
  unfolding rpad_def
  by (intro exI[of _ "(0, 0)"], simp)

adhoc_overloading rep write_cap_rep'
```

A write capability is correctly bounded by the lowest kernel storage address.

```
lemma write_cap_additional_bound[simplified, simp]:
  "snd ⌊w⌋ < unat prefix_bound" for w :: write_capability
  using write_cap_rep'[of w]
  by (auto split:prod.split)

lemma write_cap_additional_bound'[simplified, simp]:
  "unat prefix_bound ≤ n ⟹ ⌊w⌋ = (a, b) ⟹ b < n"
  using write_cap_additional_bound[of w] by simp

lemma write_cap_bound: "unat (fst ⌊w⌋) + snd ⌊w⌋ < unat prefix_bound"
  using write_cap_rep'[of w]
  by (simp split:prod.splits)
```

```
lemma write_cap_bound'[simplified, simp]: "⌊w⌋ = (a, b) ⟹ unat a + b < unat prefix_bound"
  using write_cap_bound[of w] by simp
```

There is no possible overflow in adding the number of additional addresses to the base write address.

```
lemma write_cap_no_overflow: "fst ⌊w⌋ ≤ fst ⌊w⌋ + of_nat (snd ⌊w⌋)" for w :: write_capability
  by (simp add:word_le_nat_alt unat_of_nat_eq less_imp_le)

lemma write_cap_no_overflow'[simp]: "⌊w⌋ = (a, b) ⟹ a ≤ a + of_nat b"
  for w :: write_capability
  using write_cap_no_overflow[of w] by simp
```

Auxiliary lemma: the *ith* element of the kernel address prefix is binary *1* if and only if i is smaller then the size of the prefix, otherwise it is *0*.

```
lemma nth_kern_prefix: "kern_prefix !! i = (i < size kern_prefix)"
proof−
  fix i
  {
    fix c :: nat
    assume "i < c"
    then consider "i = c − 1" | "i < c − 1 ∧ c ≥ 1"
      by fastforce
  } note elim = this
  have "i < size kern_prefix ⟹ kern_prefix !! i"
    by (subst test_bit_bl, (erule elim, simp_all)+)
  moreover have "i ≥ size kern_prefix ⟹ ¬ kern_prefix !! i" by simp
  ultimately show "kern_prefix !! i = (i < size kern_prefix)" by auto
qed
```

The *ith* bit of the lowest kernel address is *1* if and only if i is smaller or equal to the size of the kernel prefix, otherwise it is *0*.

```
lemma nth_prefix_bound[iff]:
  "prefix_bound !! i = (i ∈ {LENGTH(word32) − size (kern_prefix)..<LENGTH(word32)})"
  (is "_ = (i ∈ {?l..<?r})")
proof−
  have 0:"is_up (ucast :: 32 word ⇒ word32)" by simp
  have 1:"width (ucast kern_prefix :: word32) ≤ size kern_prefix"
    using width_ucast[of kern_prefix, OF 0] by (simp del:width_iff)
  fix i
  show "prefix_bound !! i = (i ∈ {?l..<?r})"
    using rpad_high
      [of "(ucast)(len TYPE(word32)) kern_prefix" "size (kern_prefix)" i, OF 1, simplified]
      rpad_low
      [of "(ucast)(len TYPE(word32)) kern_prefix" "size (kern_prefix)" i, OF 1, simplified]
      nth_kern_prefix[of "i − ?l", simplified] nth_ucast[of kern_prefix i, simplified]
      test_bit_size[of prefix_bound i, simplified]
  by (simp (no_asm_simp)) linarith
qed
```

Addresses from write capabilities can not contain the prefix of the kernel storage.

```
lemma write_cap_high[dest]:
  "unat a < unat prefix_bound ⟹
   ∃ i ∈ {LENGTH(word32) − size (kern_prefix)..<LENGTH(word32)}. ¬ a !! i"
  (is "_ ⟹ ∃ i ∈ {?l..<?r}. _")
  for a :: word32
proof (rule ccontr, simp del:word_size len_word ucast_bintr)
  {
    fix i
    have "(ucast kern_prefix :: word32) !! i = (i < size kern_prefix)"
      using nth_kern_prefix[of i] nth_ucast[of kern_prefix i] by auto
```

```
    moreover assume "i + ?l < ?r ⟹ a !! (i + ?l)"
    ultimately have "(a >> ?l) !! i = (ucast kern_prefix :: word32) !! i"
      using nth_shiftr[of a ?l i] by fastforce
  }
  moreover assume "∀ i∈{?l..<?r}. a !! i"
  ultimately have "a >> ?l = ucast kern_prefix" unfolding word_eq_iff using nth_ucast by auto
  moreover have "unat (a >> ?l) = unat a div 2 ^ ?l" using shiftr_div_2n' by blast
  moreover have "unat (ucast kern_prefix :: word32) = unat kern_prefix"
    by (rule unat_ucast_upcast, simp)
  ultimately have "unat a div 2 ^ ?l = unat kern_prefix" by simp
  hence "unat a ≥ unat kern_prefix * 2 ^ ?l" by simp
  hence "unat a ≥ unat prefix_bound" unfolding rpad_def by simp
  also assume "unat a < unat prefix_bound"
  finally show False ..
qed
```

High-level representation of a write capability is a set of all addresses to which the capability allows to write.

```
definition "set_of_write_cap w ≡ let (a, n) = ⌊w⌋ in {a .. a + of_nat n}" for w :: write_capability
```

```
adhoc_overloading abs set_of_write_cap
```

A write capability A is a subset of a write capability B if:

- the lowest writable address (which is the base address) of B is less than or equal to the lowest writable address of A;
- the highest writable address (which is base address plus the number of additional keys) of A is less than or equal to the highest writable address of B.

```
definition "write_cap_sub A B ≡
  let (a_A, n_A) = ⌊A⌋ in let (a_B, n_B) = ⌊B⌋ in a_B ≤ a_A ∧ a_A + of_nat n_A ≤ a_B + of_nat n_B"
  for A B :: write_capability
```

```
adhoc_overloading sub write_cap_sub
```

Prove the well-definedness assumption for the write capability format.

```
lemma write_cap_sub_iff[iff]: "a ⊆_c b = (⌈a⌉ ⊆ ⌈b⌉)" for a b :: write_capability
  unfolding write_cap_sub_def set_of_write_cap_def
  by (auto split:prod.splits)
```

```
lemmas write_cap_subsets[intro] = cap_sub.intro[OF write_cap_sub_iff]
```

Locale interpretation to prove the reflexivity and transitivity properties of a subset function of the write capability format.

```
interpretation write_cap_sub: cap_sub set_of_write_cap write_cap_sub ..
```

Low-level representation of the write capability format is a 32-byte machine word list of two elements:

- the base address;
- the number of additional addresses (also as a machine word).

```
definition "write_cap_rep w ≡ let (a, n) = ⌊w⌋ in (a, of_nat n :: word32)"
```

```
adhoc_overloading rep write_cap_rep
```

Low-level representation is injective.

```
lemma write_cap_inj[dest]: "(⌊w_1⌋ :: word32 × word32) = ⌊w_2⌋ ⟹ w_1 = w_2"
```

```
  for w1 w2 :: write_capability
  unfolding write_cap_rep_def
  by (auto
      split:prod.splits iff:write_cap_rep'_inject[symmetric]
      intro!:word_of_nat_inj simp add:rpad_def)
```

Representation function is invertible.

```
lemmas write_cap_invertible[intro] = invertible.intro[OF injI, OF write_cap_inj]

interpretation write_cap_inv: invertible write_cap_rep ..

adhoc_overloading abs write_cap_inv.inv
```

An address from the high-level representation of the write capability must be below the lowest kernel storage address.

```
lemma write_cap_prefix[dest]: "a ∈ ⌈w⌉ ⟹ ¬ limited_and prefix_bound a" for w :: write_capability
proof
  assume "a ∈ ⌈w⌉"
  hence "unat a < unat prefix_bound"
    unfolding set_of_write_cap_def
    apply (simp split:prod.splits)
    using write_cap_bound'[of w] word_less_nat_alt word_of_nat_less by fastforce
  then obtain n where "n∈{LENGTH(256 word) − size kern_prefix..<LENGTH(256 word)}" and "¬ a !! n"
    using write_cap_high[of a] by auto
  moreover assume "limited_and prefix_bound a"
  ultimately show False
    unfolding limited_and_def word_eq_iff
    by (subst (asm) nth_prefix_bound, auto)
qed
```

An address from the high-level representation is different from any address from the kernel storage.

```
lemma write_cap_safe[simp]: "a ∈ ⌈w⌉ ⟹ a ≠ ⌊a'⌋" for w :: write_capability and a' :: address
  by auto
```

3.3.3 Log capability

The log capability format includes between 0 and 4 values for log topics and 1 value that specifies the number of enforced topics. We model it as a 32-byte machine word list whose length is between 0 and 4.

```
typedef log_capability = "{ws :: word32 list. length ws ≤ 4}"
  morphisms log_cap_rep' log_capability
  by (intro exI[of _ "[]"], simp)

adhoc_overloading rep log_cap_rep'
```

High-level representation of a log capability is a set of all possible log capabilities whose list prefix in the same and equals to the given log capability.

```
definition "set_of_log_cap l ≡ {xs . prefix ⌊l⌋ xs}" for l :: log_capability

adhoc_overloading abs set_of_log_cap
```

A log capability A is a subset of a log capability B if for each log topic of B the topic is either undefined or equal to that of A. But here we specify that A is a subset of B if B is a list prefix for A. Below we prove that this conditions are equivalent.

```
definition "log_cap_sub A B ≡ prefix ⌊B⌋ ⌊A⌋" for A B :: log_capability

adhoc_overloading sub log_cap_sub
```

Prove the well-definedness assumption for the log capability format.

```
lemma log_cap_sub_iff[iff]: "a ⊆c b = (⌈a⌉ ⊆ ⌈b⌉)" for a b :: log_capability
  unfolding log_cap_sub_def set_of_log_cap_def
  by force

lemmas log_cap_subsets[intro] = cap_sub.intro[OF log_cap_sub_iff]
```

Locale interpretation to prove the reflexivity and transitivity properties of a subset function of the log capability format.

```
interpretation log_cap_sub: cap_sub set_of_log_cap log_cap_sub ..
```

Proof that that the log capability subset is defined according to the specification.

```
lemma "a ⊆c b = (∀ i < length ⌊b⌋ . ⌊a⌋ ! i = ⌊b⌋ ! i ∧ i < length ⌊a⌋)"
  (is "_ = ?R") for a b :: log_capability
  unfolding log_cap_sub_def prefix_def
proof
  let ?L = "∃ zs. ⌊a⌋ = ⌊b⌋ @ zs"
  {
    assume ?L
    moreover hence "length ⌊b⌋ ≤ length ⌊a⌋" by auto
    ultimately show "?L ⟹ ?R"
      by (auto simp add:nth_append)
  next
    assume ?R
    moreover hence len:"length ⌊b⌋ ≤ length ⌊a⌋"
      using le_def by blast
    moreover from ‹?R› have "⌊a⌋ = take (length ⌊b⌋) ⌊a⌋ @ drop (length ⌊b⌋) ⌊a⌋ "
      by simp
    moreover from ‹?R› len have "take (length ⌊b⌋) ⌊a⌋ = ⌊b⌋"
      by (metis nth_take_lemma order_refl take_all)
    ultimately show "?R ⟹ ?L" by (intro exI[of _ "drop (length ⌊b⌋) ⌊a⌋"], arith)
  }
qed
```

Low-level representation of the log capability format is a 32-byte machine word list that includes between 1 and 5 values. First value is the number of enforced topics and the rest are possible values for log topics.

```
definition "log_cap_rep l ≡ (of_nat (length ⌊l⌋) :: word32) # ⌊l⌋"

no_adhoc_overloading rep log_cap_rep'

adhoc_overloading rep log_cap_rep
```

Low-level representation is injective.

```
lemma log_cap_rep_inj[dest]: "(⌊l1⌋ :: word32 list) = ⌊l2⌋ ⟹ l1 = l2" for l1 l2 :: log_capability
  unfolding log_cap_rep_def using log_cap_rep'_inject by auto
```

Representation function is invertible.

```
lemmas log_cap_rep_invertible[intro] = invertible.intro[OF injI, OF log_cap_rep_inj]

interpretation log_cap_inv: invertible log_cap_rep ..

adhoc_overloading abs log_cap_inv.inv
```

Length of a low-level representation is correct: it is the length of the topics list plus 1 for storing the number of topics.

```
lemma log_cap_rep_length[simp]: "length ⌊l⌋ = length (log_cap_rep' l) + 1"
  unfolding log_cap_rep_def by simp
```

3.3.4 External call capability

We model the external call capability format using a record with two fields: *allow_addr* and *may_send*, with the following semantic:

- if the field *allow_addr* has value, then only the Ethereum address specified by it can be called, otherwise any address can be called. This models the *CallAny* flag and the *EthAddress* together;
- if the value of the field *may_send* is true, the any quantity of Ether can be sent, otherwise no Ether can be sent. It models the *SendValue* flag.

```
type_synonym ethereum_address = "160 word" — 20 bytes

record external_call_capability =
  allow_addr :: "ethereum_address option"
  may_send   :: bool
```

High-level representation of an external call capability is a set of all possible pairs of account addresses and Ether amount that can be sent using this capability.

```
definition "set_of_ext_cap e ≡
  {(a, v) . case_option True ((=) a) (allow_addr e) ∧ (¬ may_send e ⟶ v = (0 :: word32)) }"

adhoc_overloading abs set_of_ext_cap
```

Auxiliary abbreviation: *allow_any e* returns *True* if the field *allow_addr* of the capability *e* does not contain any value, and *False* otherwise.

```
abbreviation "allow_any e ≡ Option.is_none (allow_addr e)"
```

Auxiliary abbreviation: *the_addr e* returns the value of the field *allow_addr* of the capability *e*. It can be used only if *allow_any e* is *False*.

```
abbreviation "the_addr e ≡ the (allow_addr e)"
```

An external call capability A is a subset of an external call capability B if and only if:

- if A allows to call any Ethereum address, then B also must allow to call any address;
- if A allows to call only specified Ethereum address, then B either must allow to call any address, or it must allow to only call the same address as A;
- if A may send Ether, then B also must be able to send Ether.

```
definition "ext_cap_sub A B ≡
    (allow_any A ⟶ allow_any B)
  ∧ ((¬ allow_any A ⟶ allow_any B) ∨ (the_addr A = the_addr B))
  ∧ (may_send A ⟶ may_send B)"
  for A B :: external_call_capability

adhoc_overloading sub ext_cap_sub
```

Prove the well-definedness assumption for the external call capability format.

```
lemma ext_cap_sub_iff[iff]: "a ⊆c b = (⌈a⌉ ⊆ ⌈b⌉)" for a b :: external_call_capability
proof−
  {
    fix v' :: word32
    have "∃ v. v ≠ v'" by (intro exI[of _ "v' − 1"], simp)
  } note [intro] = this
  {
    fix a' :: ethereum_address
    have "∃ a. a ≠ a'" by (intro exI[of _ "a' − 1"], simp)
```

```
  } note [intro] = this
  show ?thesis
  unfolding set_of_ext_cap_def ext_cap_sub_def
  by (cases "allow_addr a";
      cases "allow_addr b";
      cases "may_send a";
      cases "may_send b",
      auto iff:subset_iff)
qed

lemmas ext_cap_subsets[intro] = cap_sub.intro[OF ext_cap_sub_iff]
```

Locale interpretation to prove the reflexivity and transitivity properties of a subset function of the external call capability format.

```
interpretation ext_cap_sub: cap_sub set_of_ext_cap ext_cap_sub ..
```

Helper functions to define low-level representation.

```
definition "ext_cap_val e ≡
  (of_bl ([allow_any e, may_send e]
        @ replicate 6 False) :: byte) ₁◊ case_option 0 id (allow_addr e)"

definition "ext_cap_frame e ≡
  {if allow_any e then 0 else LENGTH(ethereum_address)..<LENGTH(word32) − LENGTH(byte)}"
```

Low-level 32-byte machine word representation of the external call capability format:

- first bit is the CallAny flag;
- second bit is the SendValue flag;
- 6 undefined bits;
- 11 undefined bytes;
- 20 bytes of the Ethereum address.

```
definition "ext_cap_rep e r ≡  ext_cap_val e OR r ↑ ext_cap_frame e"
  for e :: external_call_capability

adhoc_overloading rep ext_cap_rep
```

Low-level representation is injective.

```
lemma ext_cap_rep_helper_inj[dest]: "ext_cap_val e₁ = ext_cap_val e₂ ⟹ e₁ = e₂"
  for e₁ e₂ :: external_call_capability
  unfolding ext_cap_val_def
  by (cases "allow_any e₁"; cases "allow_any e₂")
    (auto simp del:of_bl_True of_bl_False dest:word_bl.Abs_eqD split:option.splits)

lemma ext_cap_rep_helper_zero[simp]: "n ∈ ext_cap_frame e ⟹ ¬ ext_cap_val e !! n"
  unfolding ext_cap_frame_def ext_cap_val_def
  by (auto simp del:of_bl_True split:option.split)

lemma ext_cap_rep_inj[dest]: "⌊e₁⌋ r₁ = ⌊e₂⌋ r₂ ⟹ e₁ = e₂" for e₁ e₂ :: external_call_capability
proof (erule rev_mp; cases "allow_any e₁"; cases "allow_any e₂")
  let ?goal = "⌊e₁⌋ r₁ = ⌊e₂⌋ r₂ ⟶ e₁ = e₂"
  {
    {
      fix P e
      have "allow_any e ⟹ (⋀s. P (| allow_addr = None, may_send = s |)) ⟹ P e"
        by (cases e, simp add:Option.is_none_def)
```

```
  } note[elim!] = this
  note [dest] =
    restrict_inj2[of "λ s (_ :: unit). ext_cap_val (| allow_addr = None, may_send = s |)"]
  assume "allow_any e₁" and "allow_any e₂"
  thus ?goal unfolding ext_cap_rep_def by (auto simp add:ext_cap_frame_def)
next
  {
    fix P e
    have "¬ allow_any e ⟹ (⋀a s. P (| allow_addr = Some a, may_send = s |)) ⟹ P e"
      by (cases e, auto simp add:Option.is_none_def)
  } note [elim!] = this
  note [dest] = restrict_inj2[of "λ a s. ext_cap_val (| allow_addr = Some a, may_send = s |)"]
  assume "¬ allow_any e₁" and "¬ allow_any e₂"
  thus ?goal unfolding ext_cap_rep_def by (auto simp add:ext_cap_frame_def)
next
  let ?neq = "allow_any e₁ ≠ allow_any e₂"
  {
    presume ?neq
    moreover hence "msb (ext_cap_val e₁) ≠ msb (ext_cap_val e₂)"
      unfolding ext_cap_val_def msb_nth
      by (auto simp del:of_bl_True of_bl_False simp add:pad_join_high iff:test_bit_of_bl)
    ultimately show ?goal
      unfolding ext_cap_rep_def ext_cap_frame_def word_eq_iff msb_nth word_or_nth nth_restrict
      by simp  (meson less_irrefl numeral_less_iff semiring_norm(76) semiring_norm(81))
    thus ?goal .
  next
    assume "allow_any e₁" and "¬ allow_any e₂"
    thus ?neq by simp
  next
    assume "¬ allow_any e₁" and "allow_any e₂"
    thus ?neq by simp
  }
}
qed
```

Representation function is invertible.

```
lemmas ext_cap_invertible[intro] = invertible2.intro[OF inj2I, OF ext_cap_rep_inj]

interpretation ext_cap_inv: invertible2 ext_cap_rep ..

adhoc_overloading abs ext_cap_inv.inv2
```

4 Kernel state

This section contains definition of the kernel state.

4.1 Procedure data

Introduce *'a capability_list* type that is a list of capabilities of a specific type *'a*, whose length is smaller than 255.

```
typedef 'a capability_list = "{l :: 'a list. length l < 2 ^ LENGTH(byte) - 1}"
  morphisms cap_list_rep cap_list
  by (intro exI[of _ "[]"], simp)

adhoc_overloading rep cap_list_rep
```

We model a procedure using a record with the following fields:

- *eth_addr* field stores the Ethereum address of the procedure;

- *entry_cap* field is *True* if the procedure is the entry procedure, and *False* otherwise;
- other fields are lists of capabilities of corresponding types assigned to the procedure.

```
record procedure =
  eth_addr   :: ethereum_address
  call_caps  :: "prefixed_capability capability_list"
  reg_caps   :: "prefixed_capability capability_list"
  del_caps   :: "prefixed_capability capability_list"
  entry_cap  :: bool
  write_caps :: "write_capability capability_list"
  log_caps   :: "log_capability capability_list"
  ext_caps   :: "external_call_capability capability_list"

lemmas alist_simps = size_alist_def alist.Alist_inverse alist.impl_of_inverse

declare alist_simps[simp]
```

Low-level representation of the capability as it is stored in the kernel storage: given the procedure, the capability type, index and offset, it checks that all parameters are valid and correct and returns the machine word representation of the capability.

```
definition "caps_rep (k :: key) p r ty (i :: capability_index) (off :: capability_offset) ≡
  let addr = ⌊Heap_proc k (Cap ty i off)⌋ in
  case ty of
    Call ⇒ if ⌊i⌋ < length ⌊call_caps p⌋ ∧ off = 0
           then ⌊⌊call_caps p⌋ ! ⌊i⌋⌋ (r addr)
           else r addr

  | Reg  ⇒ if ⌊i⌋ < length ⌊reg_caps p⌋ ∧ off = 0
           then ⌊⌊reg_caps p⌋ ! ⌊i⌋⌋ (r addr)
           else r addr
  | Del  ⇒ if ⌊i⌋ < length ⌊del_caps p⌋ ∧ off = 0
           then ⌊⌊del_caps p⌋ ! ⌊i⌋⌋ (r addr)
           else r addr
  | Entry ⇒ r addr
  | Write ⇒ if ⌊i⌋ < length ⌊write_caps p⌋
           then
             if off = 0x00      then fst (⌊⌊write_caps p⌋ ! ⌊i⌋⌋ :: _ × word32)
             else if off = 0x01 then snd ⌊⌊write_caps p⌋ ! ⌊i⌋⌋
             else                    r addr
           else                      r addr
  | Log  ⇒ if ⌊i⌋ < length ⌊log_caps p⌋
           then
             if unat off < length ⌊⌊log_caps p⌋ ! ⌊i⌋⌋ then ⌊⌊log_caps p⌋ ! ⌊i⌋⌋ ! unat off
             else                                          r addr
           else                                            r addr
  | Send ⇒ if ⌊i⌋ < length ⌊ext_caps p⌋ ∧ off = 0
           then ⌊⌊ext_caps p⌋ ! ⌊i⌋⌋ (r addr)
           else r addr"
```

Capability representation is injective.

```
lemma caps_rep_inj[dest]:
  assumes "caps_rep k₁ p₁ r₁ = caps_rep k₂ p₂ r₂"
  shows   "length ⌊call_caps p₁⌋ = length ⌊call_caps p₂⌋   ⟹ call_caps p₁ = call_caps p₂"
    and   "length ⌊reg_caps p₁⌋ = length ⌊reg_caps p₂⌋     ⟹ reg_caps p₁ = reg_caps p₂"
    and   "length ⌊del_caps p₁⌋ = length ⌊del_caps p₂⌋     ⟹ del_caps p₁ = del_caps p₂"
    and   "length ⌊write_caps p₁⌋ = length ⌊write_caps p₂⌋ ⟹ write_caps p₁ = write_caps p₂"
    and   "length ⌊log_caps p₁⌋ = length ⌊log_caps p₂⌋     ⟹ log_caps p₁ = log_caps p₂"
    and   "length ⌊ext_caps p₁⌋ = length ⌊ext_caps p₂⌋     ⟹ ext_caps p₁ = ext_caps p₂"
proof−
```

```
from assms have eq:"⋀ ty i off. caps_rep k1 p1 r1 ty i off = caps_rep k2 p2 r2 ty i off"
  by simp
note Let_def[simp] if_splits[split] nth_equalityI[intro] cap_list_rep_inject[symmetric, iff]
{
  fix i :: nat
  let ?addr1 = "⌊Heap_proc k1 (Cap Call ⌈i⌉ 0)⌋"
  and ?addr2 = "⌊Heap_proc k2 (Cap Call ⌈i⌉ 0)⌋"
  assume idx:"i < length ⌊call_caps p1⌋"
  hence 0:"i ∈ {i. i < 2 ^ LENGTH(8 word) - 1}"
    using cap_list_rep[of "call_caps p1"] by simp
  assume "length ⌊call_caps p1⌋ = length ⌊call_caps p2⌋"
  with idx eq[of Call "⌈i⌉" 0]
  have "⌊⌊call_caps p1⌋ ! i⌋ (r1 ?addr1) = ⌊⌊call_caps p2⌋ ! i⌋ (r2 ?addr2)"
    unfolding caps_rep_def by (simp add:cap_index_inverse[OF 0])
}
thus "length ⌊call_caps p1⌋ = length ⌊call_caps p2⌋ ⟹ call_caps p1 = call_caps p2"
  by force

{
  fix i :: nat
  let ?addr1 = "⌊Heap_proc k1 (Cap Reg ⌈i⌉ 0)⌋"
  and ?addr2 = "⌊Heap_proc k2 (Cap Reg ⌈i⌉ 0)⌋"
  assume idx:"i < length ⌊reg_caps p1⌋"
  hence 0:"i ∈ {i. i < 2 ^ LENGTH(8 word) - 1}"
    using capability_list.cap_list_rep[of "reg_caps p1"] by simp
  assume "length ⌊reg_caps p1⌋ = length ⌊reg_caps p2⌋"
  with idx eq[of Reg "⌈i⌉" 0]
  have "⌊⌊reg_caps p1⌋ ! i⌋ (r1 ?addr1) = ⌊⌊reg_caps p2⌋ ! i⌋ (r2 ?addr2)"
    unfolding caps_rep_def by (simp add:cap_index_inverse[OF 0])
}
thus "length ⌊reg_caps p1⌋ = length ⌊reg_caps p2⌋ ⟹ reg_caps p1 = reg_caps p2"
  by force

{
  fix i :: nat
  let ?addr1 = "⌊Heap_proc k1 (Cap Del ⌈i⌉ 0)⌋"
  and ?addr2 = "⌊Heap_proc k2 (Cap Del ⌈i⌉ 0)⌋"
  assume idx:"i < length ⌊del_caps p1⌋"
  hence 0:"i ∈ {i. i < 2 ^ LENGTH(8 word) - 1}"
    using cap_list_rep[of "del_caps p1"] by simp
  assume "length ⌊del_caps p1⌋ = length ⌊del_caps p2⌋"
  with idx eq[of Del "⌈i⌉" 0]
  have "⌊⌊del_caps p1⌋ ! i⌋ (r1 ?addr1) = ⌊⌊del_caps p2⌋ ! i⌋ (r2 ?addr2)"
    unfolding caps_rep_def by (simp add:cap_index_inverse[OF 0])
}
thus "length ⌊del_caps p1⌋ = length ⌊del_caps p2⌋ ⟹ del_caps p1 = del_caps p2"
  by force

{
  fix i :: nat
  let ?addr1 = "⌊Heap_proc k1 (Cap Send ⌈i⌉ 0)⌋"
  and ?addr2 = "⌊Heap_proc k2 (Cap Send ⌈i⌉ 0)⌋"
  assume idx:"i < length ⌊ext_caps p1⌋"
  hence 0:"i ∈ {i. i < 2 ^ LENGTH(8 word) - 1}"
    using capability_list.cap_list_rep[of "ext_caps p1"] by simp
  assume "length ⌊ext_caps p1⌋ = length ⌊ext_caps p2⌋"
  with idx eq[of Send "⌈i⌉" 0]
  have "⌊⌊ext_caps p1⌋ ! i⌋ (r1 ?addr1) = ⌊⌊ext_caps p2⌋ ! i⌋ (r2 ?addr2)"
    unfolding caps_rep_def by (simp add:cap_index_inverse[OF 0])
}
```

```
  thus "length ⌊ext_caps p₁⌋ = length ⌊ext_caps p₂⌋ ⟹ ext_caps p₁ = ext_caps p₂"
    by force

  {
    fix i :: nat
    let ?addr₁ = "⌊Heap_proc k₁ (Cap Write ⌈i⌉ 0)⌋"
    and ?addr₂ = "⌊Heap_proc k₂ (Cap Write ⌈i⌉ 0)⌋"
    assume idx:"i < length ⌊write_caps p₁⌋"
    hence 0:"i ∈ {i. i < 2 ^ LENGTH(8 word) − 1}"
      using capability_list.cap_list_rep[of "write_caps p₁"] by simp
    assume "length ⌊write_caps p₁⌋ = length ⌊write_caps p₂⌋"
    with idx eq[of Write "⌈i⌉" "0x00"] eq[of Write "⌈i⌉" "0x01"]
    have "(⌊⌊write_caps p₁⌋ ! i⌋ :: word32 × word32) = ⌊⌊write_caps p₂⌋ ! i⌋"
      unfolding caps_rep_def by (simp add:cap_index_inverse[OF 0] prod_eqI)
  }
  thus "length ⌊write_caps p₁⌋ = length ⌊write_caps p₂⌋ ⟹ write_caps p₁ = write_caps p₂"
    by force

  {
    fix i :: nat
    let ?addr₁ = "⌊Heap_proc k₁ (Cap Log ⌈i⌉ 0)⌋"
    and ?addr₂ = "⌊Heap_proc k₂ (Cap Log ⌈i⌉ 0)⌋"
    assume idx:"i < length ⌊log_caps p₁⌋"
    hence 0:"i ∈ {i. i < 2 ^ LENGTH(8 word) − 1}"
      using capability_list.cap_list_rep[of "log_caps p₁"] by simp
    {
      fix l
      from log_cap_rep'[of l]
      have "unat (of_nat (length (log_cap_rep' l)) :: word32) = length (log_cap_rep' l)"
        by (simp add:unat_of_nat_eq)
    }
    moreover assume len:"length ⌊log_caps p₁⌋ = length ⌊log_caps p₂⌋"
    ultimately have rep_len:"length ⌊⌊log_caps p₁⌋ ! i⌋ = length ⌊⌊log_caps p₂⌋ ! i⌋"
      using idx eq[of Log "⌈i⌉" 0]
      unfolding caps_rep_def log_cap_rep_def
      by (auto simp add:cap_index_inverse[OF 0], metis)
    {
      fix off
      assume off:"off < length ⌊⌊log_caps p₁⌋ ! i⌋"
      hence "unat (of_nat off :: byte) = off"
        using log_cap_rep'[of "⌊log_caps p₁⌋ ! i"] by (simp add:unat_of_nat_eq)
      with idx off eq[of Log "⌈i⌉" "of_nat off"] len rep_len
      have "⌊⌊log_caps p₁⌋ ! i⌋ ! off = ⌊⌊log_caps p₂⌋ ! i⌋ ! off"
        unfolding caps_rep_def
        by (auto simp add:cap_index_inverse[OF 0])
    }
    with len rep_len have "⌊⌊log_caps p₁⌋ ! i⌋ = ⌊⌊log_caps p₂⌋ ! i⌋" by auto
  }
  thus "length ⌊log_caps p₁⌋ = length ⌊log_caps p₂⌋ ⟹ log_caps p₁ = log_caps p₂"
    by force
qed
```

Low-level representation of the procedure as it is stored in the kernel storage: given the procedure and the data offset it returns the machine word representation of the data that can be found by that offset.

```
definition "proc_rep k (i :: key_index) (p :: procedure) r (off :: data_offset) ≡
  let addr = ⌊off⌋ in
  let ncaps = λ n. ucast (of_nat n :: byte) OR r addr ↑ {LENGTH(byte)..<LENGTH(word32)} in
  case off of
    Addr       ⇒ ucast (eth_addr p) OR r addr ↑ {LENGTH(ethereum_address) ..<LENGTH(word32)}
```

```
| Index       ⇒ ucast ⌊i⌋ OR r addr ↑ {LENGTH(key) ..<LENGTH(word32)}
| Ncaps Call  ⇒ ncaps (length ⌊call_caps p⌋)
| Ncaps Reg   ⇒ ncaps (length ⌊reg_caps p⌋)
| Ncaps Del   ⇒ ncaps (length ⌊del_caps p⌋)
| Ncaps Entry ⇒ ncaps (of_bool (entry_cap p))
| Ncaps Write ⇒ ncaps (length ⌊write_caps p⌋)
| Ncaps Log   ⇒ ncaps (length ⌊log_caps p⌋)
| Ncaps Send  ⇒ ncaps (length ⌊ext_caps p⌋)
| Cap ty i off ⇒ caps_rep k p r ty i off"
```

Low-level representation is injective.

```
lemma restrict_ucast_inj[simplified, dest!]:
  "⟦ucast x₁ OR y₁ ↑ {l ..<LENGTH('b)} = ucast x₂ OR y₂ ↑ {l ..<LENGTH('b)};
   l = LENGTH('b); LENGTH('b) < LENGTH(word32)⟧ ⟹ x₁ = x₂"
  for x₁ x₂ :: "'b::len word" and y₁ y₂ :: word32
    by (auto dest!:restrict_inj2[of "λ x (_ :: unit). ucast x"] intro:ucast_up_inj)

lemma proc_rep_inj[dest]:
  assumes "proc_rep k₁ i₁ p₁ r₁ = proc_rep k₂ i₂ p₂ r₂"
  shows   "p₁ = p₂" and "i₁ = i₂"
proof (rule procedure.equality)
  from assms have eq:"⋀ off. proc_rep k₁ i₁ p₁ r₁ off = proc_rep k₂ i₂ p₂ r₂ off" by simp

  from eq[of Addr] show "eth_addr p₁ = eth_addr p₂"
    unfolding proc_rep_def by auto
  from eq[of Index] show "i₁ = i₂" unfolding proc_rep_def by auto

  {
    fix l :: "'b capability_list"
    from cap_list_rep[of l]
    have "unat (of_nat (length ⌊l⌋) :: byte) = length ⌊l⌋" by (simp add:unat_of_nat_eq)
  }
  hence [dest]:"⋀ l₁ :: 'b capability_list. ⋀ l₂ :: 'b capability_list.
          (of_nat (length ⌊l₁⌋) :: byte) = of_nat (length ⌊l₂⌋) ⟹ length ⌊l₁⌋ = length ⌊l₂⌋"
    by metis

  from eq[of "Cap _ _ _"] have caps:"caps_rep k₁ p₁ r₁ = caps_rep k₂ p₂ r₂"
    unfolding proc_rep_def by force

  from eq[of "Ncaps Call"] have "length ⌊call_caps p₁⌋ = length ⌊call_caps p₂⌋"
    unfolding proc_rep_def by auto
  with caps show "call_caps p₁ = call_caps p₂" ..

  from eq[of "Ncaps Reg"] have "length ⌊reg_caps p₁⌋ = length ⌊reg_caps p₂⌋"
    unfolding proc_rep_def by auto
  with caps show "reg_caps p₁ = reg_caps p₂" ..

  from eq[of "Ncaps Del"] have "length ⌊del_caps p₁⌋ = length ⌊del_caps p₂⌋"
    unfolding proc_rep_def by auto
  with caps show "del_caps p₁ = del_caps p₂" ..

  from eq[of "Ncaps Write"] have "length ⌊write_caps p₁⌋ = length ⌊write_caps p₂⌋"
    unfolding proc_rep_def by auto
  with caps show "write_caps p₁ = write_caps p₂" ..

  from eq[of "Ncaps Log"] have "length ⌊log_caps p₁⌋ = length ⌊log_caps p₂⌋"
    unfolding proc_rep_def by auto
  with caps show "log_caps p₁ = log_caps p₂" ..

  from eq[of "Ncaps Send"] have "length ⌊ext_caps p₁⌋ = length ⌊ext_caps p₂⌋"
```

```
  unfolding proc_rep_def by auto
 with caps show "ext_caps p₁ = ext_caps p₂" ..

 from eq[of "Ncaps Entry"] show "entry_cap p₁ = entry_cap p₂"
  unfolding proc_rep_def by (auto del:iffI) (simp split:if_splits add:of_bool_def)
qed simp
```

4.2 Kernel storage layout

Maximum number of procedures registered in the kernel is $2^{192} - 1$.

```
abbreviation "max_nprocs ≡ 2 ^ LENGTH(key) - 1 :: nat"
```

Introduce *procedure_list* type that is an association list of elements (a list in which each list element comprises a key and a value, and all keys are distinct), where element key is a procedure key and element value is a procedure itself.

```
typedef procedure_list = "{l :: (key, procedure) alist. size l ≤ max_nprocs}"
 morphisms proc_list_rep proc_list
 by (intro exI[of _ "Alist []"], simp)

adhoc_overloading rep proc_list_rep

adhoc_overloading rep DAList.impl_of
```

We model the kernel storage as a record with three fields:

- *curr_proc* field stores the Ethereum address of the current procedure;
- *entry_proc* field stores the Ethereum address of the entry procedure;
- *proc_list* field stores the list of all registered procedures (with their data).

```
record kernel =
 curr_proc  :: ethereum_address
 entry_proc :: ethereum_address
 proc_list  :: procedure_list
```

Here we introduce some useful abbreviations that will simplify the expression of the kernel state properties.

Number of the procedures:

```
abbreviation "nprocs σ ≡ size ⌊proc_list σ⌋"
```

Set of procedure indexes:

```
definition "proc_ids σ ≡ {0..<nprocs σ}"

abbreviation "procs σ ≡ DAList.lookup ⌊proc_list σ⌋"

definition "has_key k σ ≡ k ∈ dom (procs σ)"
```

Procedure by its key:

```
definition "proc σ k ≡ the (procs σ k)"

abbreviation "proc_key σ i ≡ fst (⌊⌊proc_list σ⌋⌋ ! i)"
```

Index of procedure:

```
definition "proc_id σ k ≡ ⌈length (takeWhile ((≠) k ∘ fst) ⌊⌊proc_list σ⌋⌋)⌉ :: key_index"

lemma proc_id_alt[simp]:
```

```
  "has_key k σ ⟹ ⌊proc_id σ k⌋ ∈ proc_ids σ"
  "has_key k σ ⟹ ⌊⌊proc_list σ⌋⌋ ! ⌊proc_id σ k⌋ = (k, proc σ k)"
proof–
  assume "has_key k σ"
  hence 0:"(k, proc σ k) ∈ set ⌊⌊proc_list σ⌋⌋"
    unfolding has_key_def proc_def DAList.lookup_def
    by auto
  hence "length (takeWhile ((≠) k ∘ fst) ⌊⌊proc_list σ⌋⌋) ∈ proc_ids σ"
    unfolding has_key_def proc_id_def proc_ids_def
    using length_takeWhile_less[of "⌊⌊proc_list σ⌋⌋ :: (key × procedure) list" "(≠) k ∘ fst"]
    by force
  moreover hence [simp]:"⌊⌈length (takeWhile ((≠) k ∘ fst) ⌊⌊proc_list σ⌋⌋)⌉ :: key_index⌋ =
                         length (takeWhile ((≠) k ∘ fst) ⌊⌊proc_list σ⌋⌋)"
    unfolding proc_ids_def
    using key_index_inverse proc_list_rep[of "proc_list σ"]
    by auto
  ultimately show 1:"⌊proc_id σ k⌋ ∈ proc_ids σ" unfolding proc_ids_def proc_id_def by simp

  from 0 have "∃! i. i < length ⌊⌊proc_list σ⌋⌋ ∧ ⌊⌊proc_list σ⌋⌋ ! i = (k, proc σ k)"
    using distinct_map by (auto intro!:distinct_Ex1)
  moreover
  {
    fix p i j
    assume 0:"i < length ⌊⌊proc_list σ⌋⌋" and 1:"j < length ⌊⌊proc_list σ⌋⌋"
    moreover assume "⌊⌊proc_list σ⌋⌋ ! i = (k, p)" and "fst (⌊⌊proc_list σ⌋⌋ ! j) = k"
    ultimately have "snd (⌊⌊proc_list σ⌋⌋ ! j) = p"
      using impl_of_distinct nth_mem distinct_map[of fst] unfolding inj_on_def
      by (metis fst_conv snd_conv)
  }
  ultimately have "∀ i < length ⌊⌊proc_list σ⌋⌋.
                   fst (⌊⌊proc_list σ⌋⌋ ! i) = k ⟶ snd (⌊⌊proc_list σ⌋⌋ ! i) = proc σ k"
    by auto
  with 1 show "⌊⌊proc_list σ⌋⌋ ! ⌊proc_id σ k⌋ = (k, proc σ k)"
    unfolding proc_id_def proc_def proc_ids_def DAList.lookup_def
    using nth_length_takeWhile[of "(≠) k ∘ fst" "⌊⌊proc_list σ⌋⌋ :: (key × procedure) list"]
    by (auto intro:prod_eqI)
qed
```

Low-level representation of the kernel storage is a 256 x 256 bits key-value store.

```
definition "kernel_rep (σ :: kernel) r a ≡
  case ⌈a⌉ of
    None          ⇒ r a
  | Some addr      ⇒ (case addr of
      Nprocs      ⇒ ucast (of_nat (nprocs σ) :: key) OR r a ↑ {LENGTH(key) ..<LENGTH(word32)}
    | Proc_key i  ⇒ ucast (proc_key σ ⌊i⌋) OR r a ↑ {LENGTH(key) ..<LENGTH(word32)}
    | Kernel      ⇒ 0
    | Curr_proc    ⇒ ucast (curr_proc σ) OR r a ↑ {LENGTH(ethereum_address) ..<LENGTH(word32)}
    | Entry_proc   ⇒ ucast (entry_proc σ) OR r a ↑ {LENGTH(ethereum_address) ..<LENGTH(word32)}
    | Heap_proc k off ⇒ if has_key k σ
                        then proc_rep k (proc_id σ k) (proc σ k) r off
                        else r a)"

adhoc_overloading rep kernel_rep

lemma proc_list_eqI[intro]:
  assumes "nprocs σ₁ = nprocs σ₂"
      and "⋀ i. i < nprocs σ₁ ⟹ proc_key σ₁ i = proc_key σ₂ i"
      and "⋀ k. ⟦has_key k σ₁; has_key k σ₂⟧ ⟹ proc σ₁ k = proc σ₂ k"
    shows "proc_list σ₁ = proc_list σ₂"
  unfolding has_key_def DAList.lookup_def proc_def
```

```
proof–
  from assms have "∀ i < nprocs σ₁.
                   snd (⌊⌊proc_list σ₁⌋⌋ ! i) = snd (⌊⌊proc_list σ₂⌋⌋ ! i)"
    unfolding has_key_def DAList.lookup_def proc_def
    apply (auto iff:fun_eq_iff)
    using
      Some_eq_map_of_iff[of "⌊⌊proc_list σ₁⌋⌋"] Some_eq_map_of_iff[of "⌊⌊proc_list σ₂⌋⌋"]
      nth_mem[of _ "⌊⌊proc_list σ₁⌋⌋"]          nth_mem[of _ "⌊⌊proc_list σ₂⌋⌋"]
      impl_of_distinct[of "⌊proc_list σ₁⌋"]     impl_of_distinct[of "⌊proc_list σ₂⌋"]
    by (metis domIff option.sel option.simps(3) surjective_pairing)
  with assms show ?thesis
    by (auto intro!:nth_equalityI prod_eqI
          iff:proc_list_rep_inject[symmetric] impl_of_inject[symmetric] fun_eq_iff)
qed
```

Low-level representation of the kernel storage is injective.

```
lemma kernel_rep_inj[dest]: "⌊σ₁⌋ r₁ = ⌊σ₂⌋ r₂ ⟹ σ₁ = σ₂" for σ₁ σ₂ :: kernel
proof (rule kernel.equality)
  assume "⌊σ₁⌋ r₁ = ⌊σ₂⌋ r₂"
  hence eq:"⋀ a. ⌊σ₁⌋ r₁ a = ⌊σ₂⌋ r₂ a" by simp

  from eq[of "⌊Curr_proc⌋"] show "curr_proc σ₁ = curr_proc σ₂"
    unfolding kernel_rep_def by auto

  from eq[of "⌊Entry_proc⌋"] show "entry_proc σ₁ = entry_proc σ₂"
    unfolding kernel_rep_def by auto

  from eq[of "⌊Nprocs⌋"] have "nprocs σ₁ = nprocs σ₂"
    unfolding kernel_rep_def
    using proc_list_rep[of "proc_list σ₁"] proc_list_rep[of "proc_list σ₂"]
    by (auto iff:of_nat_inj[symmetric])
  moreover {
    fix i
    assume "i < nprocs σ₁"
    with eq[of "⌊Proc_key ⌈i⌉⌋"] have "proc_key σ₁ i = proc_key σ₂ i"
      unfolding kernel_rep_def
      using proc_list_rep[of "proc_list σ₁"]
      by (auto simp add:key_index_inject simp add: key_index_inverse)
  }
  moreover {
    fix k
    assume "has_key k σ₁" and " has_key k σ₂"
    with eq[of "⌊Heap_proc k _⌋"] have "proc σ₁ k = proc σ₂ k"
      unfolding kernel_rep_def
      by (auto iff:fun_eq_iff[symmetric])
  }
  ultimately show "proc_list σ₁ = proc_list σ₂" ..
qed simp
```

Representation function is invertible.

```
lemmas kernel_invertible[intro] = invertible2.intro[OF inj2I, OF kernel_rep_inj]

interpretation kernel_inv: invertible2 kernel_rep ..

adhoc_overloading abs kernel_inv.inv2

lemma kernel_update_neq[simp]: "¬ limited_and prefix_bound a ⟹ ⌊σ⌋ r a = r a"
proof–
  assume "¬ limited_and prefix_bound a"
  hence "(⌈a⌉ :: address option) = None"
```

```
  using addr_prefix by − (rule ccontr, auto)
 thus ?thesis unfolding kernel_rep_def by auto
qed
```

5 Call formats

```
primrec split :: "'a::len word list ⇒ 'b::len word list list" where
  "split []       = []" |
  "split (x # xs) = word_rsplit x # split xs"

lemma cat_split: "map word_rcat (split x) = x"
  unfolding split_def
  by (induct x, simp_all add:word_rcat_rsplit)

lemma split_inj[dest]: "split x = split y ⟹ x = y"
  by (frule arg_cong[where f="map word_rcat"]) (subst (asm) cat_split)+
```

5.1 Deterministic inverse function

```
definition "maybe_inv2_tf z f l ≡
  if ∃ n. takefill z n l ∈ range2 f
  then Some (the_inv2 f (takefill z (SOME n. takefill z n l ∈ range2 f) l))
  else None"

lemma takefill_implies_prefix:
  assumes "x = takefill u n y"
  obtains (Prefix) "prefix x y" | (Postfix) "prefix y x"
proof (cases "length x ≤ length y")
  case True
  with assms have "prefix x y" unfolding takefill_alt by (simp add: take_is_prefix)
  with that show ?thesis by simp
next
  case False
  with assms have "prefix y x" unfolding takefill_alt by simp
  with that show ?thesis by simp
qed

lemma takefill_prefix_inj:
  "⟦⋀ x y. ⟦P x; P y; prefix x y⟧ ⟹ x = y; P x; P y; x = takefill u n y⟧ ⟹ x = y"
  by (elim takefill_implies_prefix) auto

definition "inj2_tf f ≡ ∀ x₁ y₁ x₂ y₂. prefix (f x₁ y₁) (f x₂ y₂) ⟶ x₁ = x₂"

lemma inj2_tfI: "(⋀ x₁ y₁ x₂ y₂. prefix (f x₁ y₁) (f x₂ y₂) ⟹ x₁ = x₂) ⟹ inj2_tf f"
  unfolding inj2_tf_def
  by blast

lemma exI2[intro]: "P x y ⟹ ∃ x y. P x y" by auto

lemma maybe_inv2_tf_inj[intro]:
  "⟦inj2_tf f; ⋀ x y y'. length (f x y) = length (f x y')⟧ ⟹ maybe_inv2_tf z f (f x y) = Some x"
  unfolding maybe_inv2_tf_def range2_def the_inv2_def inj2_tf_def
  apply (auto split:if_splits)
   apply (subst some1_equality[rotated], erule exI2)
    apply (metis length_takefill takefill_implies_prefix)
  apply (smt length_takefill takefill_implies_prefix the_equality)
  by (meson takefill_same)

lemma maybe_inv2_tf_inj':
  "⟦inj2_tf f; ⋀ x y y'. length (f x y) = length (f x y')⟧ ⟹
```

```
  maybe_inv2_tf z f v = Some x ⟹ ∃ y n. f x y = takefill z n v"
 unfolding maybe_inv2_tf_def range2_def the_inv2_def inj2_tf_def
 apply (simp split:if_splits)
 apply (subst (asm) some1_equality[rotated], erule exI2)
  apply (metis length_takefill nat_less_le not_less take_prefix take_takefill)
 by (smt prefix_order.eq_iff the1_equality)

locale invertible2_tf =
 fixes rep :: "'a ⇒ 'b ⇒ 'c::zero list" ("⌊_⌋")
 assumes inj:"inj2_tf rep"
    and len_inv:"⋀ x y y'. length (rep x y) = length (rep x y')"
begin
definition inv2_tf :: "'c list ⇒ 'a option" where "inv2_tf ≡ maybe_inv2_tf 0 rep"

lemmas inv2_tf_inj[folded inv2_tf_def, simp] = maybe_inv2_tf_inj[where z=0, OF inj len_inv]

lemmas inv2_tf_inj'[folded inv2_tf_def, dest] = maybe_inv2_tf_inj'[where z=0, OF inj len_inv]
end
```

5.2 Register system call

```
definition "wf_cap c l ≡
 case (c, l) of
   (Call,  [c])      ⇒ (⌈c⌉ :: prefixed_capability option) ≠ None
 | (Reg,   [c])      ⇒ (⌈c⌉ :: prefixed_capability option) ≠ None
 | (Del,   [c])      ⇒ (⌈c⌉ :: prefixed_capability option) ≠ None
 | (Entry, [])       ⇒ True
 | (Write, [c1, c2]) ⇒ (⌈(c1, c2)⌉ :: write_capability option) ≠ None
 | (Log,   c)        ⇒ (⌈c⌉ :: log_capability option) ≠ None
 | (Send,  [c])      ⇒ (⌈c⌉ :: external_call_capability option) ≠ None
 | _                 ⇒ False"

lemma length_wf_cap[dest]: "wf_cap c l ⟹ length l ≤ 5"
 unfolding wf_cap_def using log_cap_rep'
 by (auto split:capability.splits list.splits)

definition "same_cap c l₁ l₂ ≡
 case (c, l₁, l₂) of
   (Call,  [c₁], [c₂])          ⇒ the ⌈c₁⌉ = (the ⌈c₂⌉ :: prefixed_capability)
 | (Reg,   [c₁], [c₂])          ⇒ the ⌈c₁⌉ = (the ⌈c₂⌉ :: prefixed_capability)
 | (Del,   [c₁], [c₂])          ⇒ the ⌈c₁⌉ = (the ⌈c₂⌉ :: prefixed_capability)
 | (Entry, [],   [])            ⇒ True
 | (Write, [c1₁, c2₁], [c1₂, c2₂]) ⇒ the ⌈(c1₁, c2₁)⌉ = (the ⌈(c1₂, c2₂)⌉ :: write_capability)
 | (Log,   c₁,   c₂)              ⇒ the ⌈c₁⌉ = (the ⌈c₂⌉ :: log_capability)
 | (Send,  [c₁], [c₂])            ⇒ the ⌈c₁⌉ = (the ⌈c₂⌉ :: external_call_capability)
 | _                            ⇒ False"

definition "overwrite_cap c l r ≡
 case (c, l) of
   (Call,  [c])      ⇒ [⌊the ⌈c⌉ :: prefixed_capability⌋ (r ! 0)]
 | (Reg,   [c])      ⇒ [⌊the ⌈c⌉ :: prefixed_capability⌋ (r ! 0)]
 | (Del,   [c])      ⇒ [⌊the ⌈c⌉ :: prefixed_capability⌋ (r ! 0)]
 | (Entry, [])       ⇒ []
 | (Write, [c1, c2]) ⇒ let (c1, c2) = ⌊the ⌈(c1, c2)⌉ :: write_capability⌋ in [c1, c2]
                        — for mere consistency, no actual need in this, can be just [c1, c2]
 | (Log,   c)        ⇒ ⌊the ⌈c⌉ :: log_capability⌋
 | (Send,  [c])      ⇒ [⌊the ⌈c⌉ :: external_call_capability⌋ (r ! 0)]"

lemma overwrite_cap_wf: "wf_cap c l ⟹ wf_cap c (overwrite_cap c l r)"
 unfolding wf_cap_def overwrite_cap_def
 by (auto split:capability.splits list.splits simp add:write_cap_inv.inv_inj')
```

```
abbreviation "zero_fill l ≡ replicate (length l) 0"

lemma same_cap_inj[dest]:
  "same_cap c l₁ l₂ ⟹ overwrite_cap c l₁ (zero_fill l₁) = overwrite_cap c l₂ (zero_fill l₂)"
  unfolding same_cap_def overwrite_cap_def
  by (simp split:capability.splits)
     (auto split:capability.splits list.splits)+

lemma overwrite_cap_inj[dest]:
  "⟦overwrite_cap c l₁ r₁ = overwrite_cap c l₂ r₂; wf_cap c l₁; wf_cap c l₂⟧ ⟹ same_cap c l₁ l₂"
  unfolding wf_cap_def overwrite_cap_def same_cap_def
  by (simp split:capability.splits)
     (auto split:capability.splits list.splits simp add:write_cap_inv.inv_inj')

lemma length_overwrite_cap[simp]: "wf_cap c l ⟹ length (overwrite_cap c l r) = length l"
  unfolding wf_cap_def overwrite_cap_def
  by (auto split:capability.splits list.split prod.split)

typedef capability_data =
  "{ l :: ((capability × capability_index) × word32 list) list.
      ∀ ((c, _), l) ∈ set l. wf_cap c l ∧ l = overwrite_cap c l (replicate (length l) 0) }"
  morphisms cap_data_rep' cap_data
  by (intro exI[of _ "[]"], simp)

adhoc_overloading rep cap_data_rep'

adhoc_overloading abs cap_data

record register_call_data =
  proc_key :: key
  eth_addr :: ethereum_address
  cap_data :: capability_data

no_adhoc_overloading rep cap_index_rep

no_adhoc_overloading abs cap_index_inv.inv

definition "cap_index_rep0 i ≡ of_nat ⌊i⌋ :: byte" for i :: capability_index

adhoc_overloading rep cap_index_rep0

lemma width_cap_index0: "width ⌊i⌋ ≤ LENGTH(byte)" for i :: capability_index by simp

lemma width_cap_index0'[simp]: "LENGTH(byte) ≤ n ⟹ width ⌊i⌋ ≤ n"
  for i :: capability_index by simp

lemma cap_index_inj0[simp]: "(⌊i₁⌋ :: byte) = ⌊i₂⌋ ⟹ i₁ = i₂" for i₁ i₂ :: capability_index
  unfolding cap_index_rep0_def
  using cap_index_rep'[of i₁] cap_index_rep'[of i₂] word_of_nat_inj[of "⌊i₁⌋" "⌊i₂⌋"]
        cap_index_rep'_inject
  by force

lemmas cap_index0_invertible[intro] = invertible.intro[OF injI, OF cap_index_inj0]

interpretation cap_index_inv0: invertible cap_index_rep0 ..

adhoc_overloading abs cap_index_inv0.inv

abbreviation "cap_data_rep_single r (c :: capability) (i :: capability_index) l j d ≡
```

```
  [ucast (of_nat (3 + length l) :: byte) OR (r ! j) ↑ {LENGTH(byte) ..<LENGTH(word32)},
   ucast ⌊c⌋ OR (r ! (j + 1)) ↑ {LENGTH(byte) ..<LENGTH(word32)},
   ucast ⌊i⌋ OR (r ! (j + 2)) ↑ {LENGTH(byte) ..<LENGTH(word32)}]
  @ overwrite_cap c l (drop (j + 3) r)"

definition "cap_data_rep0 r ≡
  λ ((c, i), l) (j, d). (j + 3 + length l, cap_data_rep_single r c i l j d # d)"

lemma length_cap_data_rep0:
  fixes d :: capability_data
  assumes "cap_data_rep0 r ((c, i), l) acc = (j, x # xs)" and "((c, i), l) ∈ set ⌊d⌋"
  shows   "length x = unat (hd x AND mask LENGTH(byte))"
proof−
  from assms(2) have "wf_cap c l" using cap_data_rep'[of d] by auto
  with assms(1) show ?thesis
    unfolding cap_data_rep0_def
    by (auto split:prod.splits simp add:unat_ucast_upcast unat_of_nat_eq)
      (auto simp add:unat_of_nat_eq)
qed

lemma length_cap_data_rep0':
  "⟦[l] = snd (cap_data_rep0 r x acc); x ∈ set ⌊d⌋⟧ ⟹
   length l = unat (hd l AND mask LENGTH(byte))"
  (is "⟦?l; ?in_set⟧ ⟹ _")
  for d :: capability_data
proof−
  assume ?l and ?in_set
  obtain c i l' j ls
    where "cap_data_rep0 r ((c, i), l') acc = (j, l # ls)"
      and "((c, i), l') ∈ set ⌊d⌋"
  proof (cases "cap_data_rep0 r x acc")
    fix j lls
    assume 0:"cap_data_rep0 r x acc = (j, lls)"
    with ‹?l› have 1:"lls = l # []" by (metis snd_conv)
    from ‹?l› show ?thesis proof (cases x)
      fix ci l'
      assume 2:"x = (ci, l')"
      show ?thesis proof (cases ci)
        fix c i
        assume "ci = (c, i)"
        with that[of c i l' j "[]"] 0 1 2 ‹?in_set› ‹?l› show ?thesis by simp
      qed
    qed
  qed
  thus ?thesis using length_cap_data_rep0 by simp
qed

definition "cap_data_rep (d :: capability_data) r ≡ fold (cap_data_rep0 r) ⌊d⌋"

lemma cap_data_rep'_tail: "⌊d⌋ = x # xs ⟹ xs = ⌊⌈xs⌉⌋" for d :: capability_data
  using cap_data_rep'[of d]
  by (auto intro:cap_data_inverse[symmetric])

lemma length_snd_cap_data_rep[simp]:
  "length (snd (cap_data_rep d r i)) = length ⌊d⌋ + length (snd i)"
proof (induction "⌊d⌋" arbitrary:d)
  case Nil
  thus ?case unfolding cap_data_rep_def by simp
next
  {
```

```
    fix r xs i
    have "length (snd (fold (cap_data_rep0 r) xs i)) = length xs + length (snd i)"
      unfolding cap_data_rep0_def by (induction xs arbitrary: i, simp_all split:prod.split)
  } note [simp] = this
  case (Cons x xs)
  from Cons.hyps(1)[of "⌈xs⌉"] show ?case
    unfolding cap_data_rep_def  using cap_data_rep'_tail
    by ((subst Cons.hyps(2)[symmetric])+, simp split:prod.splits)
      (simp add:cap_data_rep0_def split:prod.splits)
qed

lemma cap_data_rep_inj[dest]:
  "⟦cap_data_rep d1 r1 i1 = cap_data_rep d2 r2 i2; length (snd i1) = length (snd i2)⟧ ⟹ d1 = d2"
  (is "⟦?eq_rep d1 i1 d2 i2; ?eq_length i1 i2⟧ ⟹ _")
proof (induction "⌊d1⌋" arbitrary:d1 d2 i1 i2)
  case Nil
  moreover hence "length (snd (cap_data_rep d1 r1 i1)) = length (snd i1)" by (simp (no_asm))
  ultimately have "⌊d1⌋ = ⌊d2⌋" by simp
  thus ?case by (simp add:cap_data_rep'_inject)
next
  {
    fix xs j1 j2 l1 l2
    have "fold (cap_data_rep0 r1) xs (j1, l1) = fold (cap_data_rep0 r2) xs (j2, l2) ⟹ l1 = l2"
      unfolding cap_data_rep0_def
      by (induction xs arbitrary: j1 j2 l1 l2, auto split:prod.splits)
  } note inj = this
  case (Cons x xs)
  hence "length ⌊d2⌋ = length ⌊d1⌋" by (metis add_right_cancel length_snd_cap_data_rep)
  with ‹x # xs = ⌊d1⌋› obtain y ys where "⌊d2⌋ = y # ys" by (metis length_Suc_conv)
  from ‹x # xs = ⌊d1⌋› have d1:"⌊d1⌋ = x # xs" ..
  note d2 = ‹⌊d2⌋ = y # ys›
  from ‹?eq_rep d1 i1 d2 i2› obtain i1' and i2'
    where "cap_data_rep ⌈xs⌉ r1 i1' = cap_data_rep ⌈ys⌉ r2 i2'"
      and   "length (snd i1') = length (snd i1) + 1"
      and   "length (snd i2') = length (snd i2) + 1"
    unfolding cap_data_rep_def cap_data_rep0_def
    using cap_data_rep'_tail[OF d2] cap_data_rep'_tail[OF d1]
    by (auto simp add:d1 d2 split:prod.split)
  with ‹?eq_rep d1 i1 d2 i2› ‹?eq_length i1 i2› have tls:"xs = ys"
    using cap_data_rep'_tail[OF d1] cap_data_rep'_tail[OF d2]
    by (auto dest:Cons.hyps(1)[OF cap_data_rep'_tail[OF d1]])
  with ‹?eq_rep d1 i1 d2 i2› d1 d2 have "snd (cap_data_rep0 r1 x i1) = snd (cap_data_rep0 r2 y i2)"
    unfolding cap_data_rep_def
    by auto (metis inj prod.collapse)
  moreover have "wf_cap (fst (fst x)) (snd x)" and  "wf_cap (fst (fst y)) (snd y)"
    using cap_data_rep'[of d1] d1 cap_data_rep'[of d2]  d2
   by auto
  ultimately have "x = y" unfolding cap_data_rep0_def
    apply (auto split:prod.splits
        del:cap_type_rep_inj overwrite_cap_inj
        dest!:cap_type_rep_inj overwrite_cap_inj)
    using cap_data_rep'[of d1] d1 cap_data_rep'[of d2] d2
    by auto
  with tls d1 d2 have "⌊d1⌋ = ⌊d2⌋" by simp
  thus ?case by (simp add:cap_data_rep'_inject)
qed

lemma cap_data_rep_lengths:
  "list_all ((≠) []) l ⟹ list_all ((≠) []) (snd (cap_data_rep d r (i, l)))"
proof (induction "⌊d⌋" arbitrary:d i l)
```

```
  case Nil
  thus ?case unfolding cap_data_rep_def by simp
next
  case (Cons x xs)
  then obtain i' l' where "cap_data_rep0 r x (i, l) = (i', l')" and "list_all ((≠) []) l'"
    unfolding cap_data_rep0_def by (induction x) auto
  with Cons show ?case
    using cap_data_rep'_tail[of d, OF Cons.hyps(2)[symmetric]] Cons.hyps(1)[of "⌈xs⌉" l' i']
    unfolding cap_data_rep_def
    by (rewrite in ‹_ # _ = ⌊d⌋› in asm eq_commute) auto
qed

lemma cap_data_rep_index[simp]:
  assumes "sum_list (map length l) ≤ i"
  shows "fst (cap_data_rep d r (i, l)) =
        sum_list (map length (snd (cap_data_rep d r (i, l)))) + (i − sum_list (map length l))"
  using assms
proof (induction "⌊d⌋" arbitrary:d i l)
  case Nil
  thus ?case unfolding cap_data_rep_def by auto
next
  case (Cons x xs)
  from Cons(2) have wf:"wf_cap (fst (fst x)) (snd x)"
    using cap_data_rep'[of d] list.set_intros(1)[of x xs]
    by (induction x) auto
  hence 0:"length (overwrite_cap (fst (fst x)) (snd x) (drop (i + 3) r)) = length (snd x)" by simp
  let "?i'" = "fst (cap_data_rep0 r x (i, l))"
    and "?l'" = "snd (cap_data_rep0 r x (i, l))"
  from 0 have "sum_list (map length ?l') = sum_list (map length l) + length (snd x) + 3"
    unfolding cap_data_rep0_def by (auto split:prod.splits)
  hence 1:"?i' = sum_list (map length ?l') + (i − sum_list (map length l))"
    unfolding cap_data_rep0_def using Cons(3) by (simp split:prod.splits)
  from Cons(3) have 2:"sum_list (map length ?l') ≤ ?i'"
    unfolding cap_data_rep0_def using wf by (auto split:prod.splits)
  from Cons(1)[of "⌈xs⌉" ?l' ?i', OF _ 2] cap_data_rep'_tail[OF Cons(2)[symmetric]]
  show ?case unfolding cap_data_rep_def by ((subst Cons(2)[symmetric])+, simp) (insert 1, simp)
qed

lemma cap_data_rep_dest:
  assumes "snd (cap_data_rep d r (i, [])) ≠ []"
  obtains i' where
    "snd (cap_data_rep d r (i, l)) =
   hd (snd (cap_data_rep0 r (last ⌊d⌋) (i', []))) # snd (cap_data_rep ⌈butlast ⌊d⌋⌉ r (i, l))"
  using assms(1)
proof (induction "⌊d⌋" arbitrary:d i l ?thesis)
  case Nil
  thus ?case unfolding cap_data_rep_def by simp
next
  case nonemp:(Cons x xs)
  show ?case proof (cases xs)
    case Nil
    from nonemp(1,3,4) show ?thesis
      unfolding cap_data_rep_def cap_data_rep0_def using cap_data_inverse
      by (simp add:nonemp(2)[symmetric] Nil split:prod.splits)
  next
    case (Cons x' xs')
    let ?l' = "snd (cap_data_rep0 r x (i, l))"
      and ?i' = "fst (cap_data_rep0 r x (i, l))"
    from cap_data_rep'_tail[OF nonemp(2)[symmetric]] have xs:"⌊⌈xs⌉⌋ = xs" ..
    let ?repx' = "cap_data_rep0 r x' (?i', [])"
```

```
    have lenx':"length (snd ?repx') > 0" unfolding cap_data_rep0_def by (simp split:prod.split)
    from cap_data_rep'_tail[of "⌈xs⌉"] xs Cons have xs':"⌊⌈xs'⌉⌋ = xs'" by simp
    from xs' have "⋀ i l. length l ≤ length (snd (cap_data_rep ⌈xs'⌉ r (i, l)))"
    proof (induction xs')
      case Nil
      thus ?case by simp
    next
      case (Cons y ys)
      let ?i' = "fst (cap_data_rep0 r y (i, l))"
        and ?l' = "snd (cap_data_rep0 r y (i, l))"
      note 0 = cap_data_rep'_tail[OF Cons(2), symmetric]
      with Cons(1)[OF 0, of ?l' ?i'] Cons(2)
      show ?case unfolding cap_data_rep_def cap_data_rep0_def by (simp split:prod.splits)
    qed
    from this[of "snd ?repx'" "fst ?repx'"] xs xs' Cons lenx'
    have 0:"snd (cap_data_rep ⌈x' # xs'⌉ r (?i', [])) ≠ []" unfolding cap_data_rep_def by auto
    from nonemp(2) Cons last_ConsR[of xs x] have 1:"last xs = last ⌊d⌋" by simp
    from cap_data_inverse[of "butlast xs"] cap_data_rep'[of "⌈xs⌉"] xs
    have 2:"⌊⌈butlast xs⌉⌋ = butlast xs" by (auto split:prod.splits dest!:in_set_butlastD)
    from cap_data_inverse[of "butlast ⌊d⌋"] cap_data_rep'[of "d"]
    have 3:"⌊⌈butlast ⌊d⌋⌉⌋ = butlast ⌊d⌋" by (auto split:prod.splits dest!:in_set_butlastD)
    from Cons have 4:"butlast ⌊d⌋ = x # butlast xs" by (rewrite nonemp(2)[symmetric], simp)
    from nonemp(1)[of "⌈xs⌉" ?i' ?l', OF xs[symmetric]] 0 Cons obtain i'' where
      "snd (cap_data_rep ⌈xs⌉ r (?i', ?l')) =
         hd (snd (cap_data_rep0 r (last xs) (i'', []))) #
           snd (cap_data_rep ⌈butlast xs⌉ r (?i', ?l'))"
      using xs
      by auto
    with nonemp(3) xs show ?thesis unfolding cap_data_rep_def
      by (rewrite in asm nonemp(2)[symmetric]) (rewrite in asm 3, simp add: 1 2 4)
  qed
qed

definition "cap_data_rep1 r ≡
  λ ((c, i), l) (j, d). (j + 3 + length l, d @ [cap_data_rep_single r c i l j d])"

lemma cap_data_rep1_fold_pull[simp]:
  "snd (fold (cap_data_rep1 r) d (i, x # xs)) = x # snd (fold (cap_data_rep1 r) d (i, xs))"
proof (induction d arbitrary:xs i)
  case Nil
  thus ?case by simp
next
  case (Cons d ds)
  obtain xs' i' where
    "cap_data_rep1 r d (i, x # xs) = (i', x # xs @ xs')" and
    "cap_data_rep1 r d (i, xs) = (i', xs @ xs')"
    unfolding cap_data_rep1_def by (induction d) auto
  with Cons(1)[of i' "xs @ xs'"] show ?case by simp
qed

lemma cap_data_rep_rel:
  "rev (snd (cap_data_rep d r (i, l))) = rev l @ snd (fold (cap_data_rep1 r) ⌊d⌋ (i, []))"
proof (induction "⌊d⌋" arbitrary: d i l)
  case Nil
  thus ?case unfolding cap_data_rep_def by simp
next
  case (Cons x xs)
  from cap_data_rep'_tail[OF Cons(2)[symmetric]] have xs:"⌊⌈xs⌉⌋ = xs" ..
  let ?i' = "fst (cap_data_rep0 r x (i, l))"
    and ?l' = "snd (cap_data_rep0 r x (i, l))"
```

```
  obtain i'' x' where 0:"cap_data_rep1 r x (i, []) = (i'', x' # [])"
    unfolding cap_data_rep1_def by (induction x) auto
  hence 1:"rev (snd (cap_data_rep0 r x (i, []))) = [x']"
    unfolding cap_data_rep0_def cap_data_rep1_def by (induction x) auto
  have [simp]: "fst (cap_data_rep0 r x (i, [])) = fst (cap_data_rep1 r x (i, []))"
    unfolding cap_data_rep0_def cap_data_rep1_def by (induction x) auto
  have [simp]:
    "cap_data_rep0 r x (i, l) =
    (fst (cap_data_rep0 r x (i, [])), snd (cap_data_rep0 r x (i, [])) @ l)"
    unfolding cap_data_rep0_def by (simp split:prod.split)
  from Cons(1)[of "⌈xs⌉" ?i' ?l', OF xs[symmetric]] xs
  show ?case unfolding cap_data_rep_def
    by (simp add: Cons(2)[symmetric] 0 1)
qed

lemma concat_cap_data_rep_inj_snd[dest]:
  fixes d1' d2' :: capability_data
  assumes "concat (snd (fold (cap_data_rep1 r1) d1 (i1, []))) =
           concat (snd (fold (cap_data_rep1 r2) d2 (i2, [])))"
  assumes "d1 = ⌊d1'⌋" and "d2 = ⌊d2'⌋"
  shows   "snd (fold (cap_data_rep1 r1) d1 (i1, [])) =
           snd (fold (cap_data_rep1 r2) d2 (i2, []))"
  using assms
proof (induction d1 arbitrary: d1' d2 d2' i1 i2)
  case Nil
  from Nil(3) have 0: "snd (fold (cap_data_rep1 r2) d2 (i2, [])) =
                       rev (snd (cap_data_rep d2' r2 (i2, [])))"
    by (subst rev_is_rev_conv[symmetric], simp add:cap_data_rep_rel)
  from Nil(3) have 1:"d2 ≠ [] ⟹ set (snd (cap_data_rep d2' r2 (i2, []))) ≠ {}"
    using length_snd_cap_data_rep[of d2' r2 "(i2, [])"] by force
  from Nil[simplified] have "d2 ≠ [] ⟹ False"
    using cap_data_rep_lengths[of "[]" d2' r2 i2, simplified, unfolded list_all_def]
    by (subst (asm) 0) (subst (asm) set_rev, frule 1, metis equals0I)
  thus ?case by (cases d2, simp_all)
next
  case (Cons x xs)
  obtain i1' l1' where
    0:"cap_data_rep1 r1 x (i1, []) = (i1', l1' # [])" and
    1:"l1' ≠ []" and
    2:"⌈l1'⌉ = snd (cap_data_rep1 r1 x (i1, []))"
    unfolding cap_data_rep1_def by (induction x) auto
  have
    l:"concat (snd (fold (cap_data_rep1 r1) (x # xs) (i1, []))) =
     l1' @ concat (snd (fold (cap_data_rep1 r1) xs (i1', [])))"
    by (simp add:0)
  from Cons(2) have "snd (fold (cap_data_rep1 r2) d2 (i2, [])) ≠ []"
    by (auto simp add:0 1)
  hence "d2 ≠ []" by auto
  then obtain y ys where 3:"d2 = y # ys" by (cases d2, auto)
  obtain i2' l2' where
    4:"cap_data_rep1 r2 y (i2, []) = (i2', l2' # [])" and
    5:"l2' ≠ []" and
    6:"⌈l2'⌉ = snd (cap_data_rep1 r2 y (i2, []))"
    unfolding cap_data_rep1_def by (induction y) auto
  have
    r:"concat (snd (fold (cap_data_rep1 r2) d2 (i2, []))) =
       l2' @ concat (snd (fold (cap_data_rep1 r2) ys (i2', [])))"
    by (simp add: 3 4)

  from 2 have 7:"⌈l1'⌉ = snd (cap_data_rep0 r1 x (i1, []))"
```

```
    unfolding cap_data_rep0_def cap_data_rep1_def by (cases x) auto
  from Cons(3) have 8:"x ∈ set ⌊d1'⌋" using list.set_intros(1)[of x xs] by simp
  note 9 = length_cap_data_rep0'[OF 7 8]
  from 6 have 10:"⌈l2'⌉ = snd (cap_data_rep0 r2 y (i2, []))"
    unfolding cap_data_rep0_def cap_data_rep1_def by (cases y) auto
  from Cons(4) 3 have 11:"y ∈ set ⌊d2'⌋" using list.set_intros(1)[of y ys] by simp
  note 12 = length_cap_data_rep0'[OF 10 11]
  from Cons(2) l r 1 5 9 12 have 13:"l1' = l2'" by (metis append_eq_append_conv hd_append2)
  with Cons(2) l r
  have 14:"concat (snd (fold (cap_data_rep1 r1) xs (i1', []))) =
          concat (snd (fold (cap_data_rep1 r2) ys (i2', [])))" by simp

  note xs = cap_data_rep'_tail[OF Cons(3)[symmetric]]
  from cap_data_rep'_tail[of d2'] Cons(4) 3 have ys:"ys = ⌊⌈ys⌉⌋" by blast
  note 15 = Cons(1)[OF 14 xs ys]

  from 0 3 4 13 15 show ?case by simp
qed

lemma concat_cap_data_rep_inj[simplified, dest]:
  "(concat ∘ rev ∘ snd) (cap_data_rep d1 r1 (i, [])) =
   (concat ∘ rev ∘ snd) (cap_data_rep d2 r2 (i, [])) ⟹
   cap_data_rep d1 r1 (i, []) = cap_data_rep d2 r2 (i, [])"
  (is "?prem ⟹ _")
proof
  assume ?prem
  hence
    "concat (snd (fold (cap_data_rep1 r1) ⌊d1⌋ (i, []))) =
     concat (snd (fold (cap_data_rep1 r2) ⌊d2⌋ (i, [])))"
    by (simp add:cap_data_rep_rel)
  hence "snd (fold (cap_data_rep1 r1) ⌊d1⌋ (i, [])) = snd (fold (cap_data_rep1 r2) ⌊d2⌋ (i, []))"
    by auto
  thus "snd (cap_data_rep d1 r1 (i, [])) = snd (cap_data_rep d2 r2 (i, []))"
    by (simp add:cap_data_rep_rel[where l="[]", simplified, symmetric])
  thus "fst (cap_data_rep d1 r1 (i, [])) = fst (cap_data_rep d2 r2 (i, []))"
    by simp
qed

definition "reg_call_rep d r ≡
    [ucast (proc_key d) OR (r ! 0) ↑ {LENGTH(key) ..<LENGTH(word32)},
     ucast (eth_addr d) OR (r ! 1) ↑ {LENGTH(ethereum_address) ..<LENGTH(word32)}] @
     ((concat ∘ rev ∘ snd) (cap_data_rep (cap_data d) r (2, [])))"

adhoc_overloading rep reg_call_rep

lemma reg_call_rep_inj[dest]: "⌊d1⌋ r1 = ⌊d2⌋ r2 ⟹ d1 = d2" for d1 d2 :: register_call_data
proof (rule register_call_data.equality)
  assume eq:"⌊d1⌋ r1 = ⌊d2⌋ r2"

  from eq show "proc_key d1 = proc_key d2" unfolding reg_call_rep_def by auto
  from eq show "eth_addr d1 = eth_addr d2" unfolding reg_call_rep_def by auto

  from eq show "cap_data d1 = cap_data d2" unfolding reg_call_rep_def by auto
qed simp

datatype result =
    Success storage
  | Revert

abbreviation "SYSCALL_NOEXIST ≡ 0xaa"
```

```
abbreviation "SYSCALL_BADCAP ≡ 0x33"

definition "cap_type_opt_rep c ≡ case c of Some c ⇒ ⌊c⌋ | None ⇒ 0x00"
  for c :: "capability option"

adhoc_overloading rep cap_type_opt_rep

lemma cap_type_opt_rep_inj[intro]: "inj cap_type_opt_rep" unfolding cap_type_opt_rep_def inj_def
  by (auto split:option.split)

lemmas cap_type_opt_invertible[intro] = invertible.intro[OF cap_type_opt_rep_inj]

interpretation cap_type_opt_inv: invertible cap_type_opt_rep ..

adhoc_overloading abs cap_type_opt_inv.inv

definition call :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "call _ _ s ≡ (Success s, [])"

definition register :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "register _ _ s ≡ (Success s, [])"

definition delete :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "delete _ _ s ≡ (Success s, [])"

definition set_entry :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "set_entry _ _ s ≡ (Success s, [])"

definition write_addr :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "write_addr _ _ s ≡ (Success s, [])"

definition log :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "log _ _ s ≡ (Success s, [])"

definition external :: "capability_index ⇒ byte list ⇒ storage ⇒ result × byte list" where
  "external _ _ s ≡ (Success s, [])"

definition execute :: "byte list ⇒ storage ⇒ result × byte list" where
  "execute c s ≡ case takefill 0x00 2 c of ct # ci # c ⇒
    (case ⌈ct⌉ of
       None          ⇒ (Revert, [SYSCALL_NOEXIST])
     | Some None     ⇒ (Success s, [])
     | Some (Some ct) ⇒ (case ⌈ci⌉ of
         None        ⇒ (Revert, [SYSCALL_BADCAP]) — Capability index out of bounds
       | Some ci     ⇒ (case ct of
           Call      ⇒ call ci c s
         | Reg       ⇒ register ci c s
         | Del       ⇒ delete ci c s
         | Entry     ⇒ set_entry ci c s
         | Write     ⇒ write_addr ci c s
         | Log       ⇒ log ci c s
         | Send      ⇒ external ci c s)))"


end
```